
3. Minimum-Cost Input Models

The diet formulation in our introductory chapters is only one representative
of a broad class of input optimization models. We now give a more general
definition of the minimum-cost input selection problem and the associated op-
timization model. We observe that this model corresponds closely to the diet
model introduced in the preceding chapters, but that other interpretations of
the inputs and outputs give rise to useful blending, scheduling, cutting, and
packing models.

3.1 The input optimization problem

Imagine a process of some sort that combines a variety of inputs. From each
unit of a particular input, the process derives fixed amounts of each of several
outputs, with the amount of each output produced being proportional to the
amount of each input consumed. Each input also has a cost per unit, again
proportional to the amount. Our goal is to choose an amount of each input to
be purchased, within specified limits, so that the total outputs thereby produced
meet specified requirements at the least total cost.

We build a mathematical model of this problem (shown in Figure 3–1) much
as we did for the diet problem in Chapter 1. We start by defining the funda-
mental sets of entities: inputs I and outputs O. We then say that each unit of
an input i 2 I can be obtained at a cost ci, and contributes amounts aoi of the
outputs o 2 O. The number of units of input i 2 I consumed must lie between
li and ui, while the number of units of output o 2 O produced must be at least
bo and at most do.

The decision variables are the amounts of the inputs to be purchased, so we
may write them as xi for each i 2 I . Then since the cost of input i purchased is
proportional to the amount of input i purchased, the cost is given by cixi, and
the objective is to minimize the total of these costs,

P
i2I cixi. Similarly since

the amount of output o derived from input i is proportional to the amount of
o purchased, the amount of o derived from input i is given by aoixi. The total
of output o derived from all inputs is thus

P
i2I aoixi, and the constraints say

that this amount lies between the requirements,

bo �
P
i2I aoixi � do;

for each output o 2 O.
An analogous description in the AMPL modeling language is given by Figure

3–2. The transcription from mathematics to AMPL here proceeds in the same
general way as it did for the diet model in Chapter 1 (from Figure 1–3 to Figure
1–4). The set J is denoted INPUTS, the variables xi become InBuy[i], and the
limits li and ui become inLo[i] and inHi[i]. The conversion of the variable
declaration is thus

li � xi � ui, for each i 2 I -!
var InBuy {i in INPUTS} >= inLo[i], <= inHi[i]

The costs ci become inCost[i], so that the expression for the objective is con-
verted as

P
i2I cixi -! sum {i in INPUTS} inCost[i] * InBuy[i]

This expression is named TotalCost and is placed in the minimize declaration.
It remains to denote the set of outputs O as OUTPUTS, the output require-

ments bo and do as outLo[o] and outHi[o], and the “input-output coefficients”

Given I , a set of inputs

O, a set of outputs

and aoi � 0, for each o 2 O and i 2 I :
the units of output o provided by one unit of input i
do � bo � 0, for each o 2 O:
bounds on the amount of output o produced

ci � 0, for each i 2 I :
the cost per unit of input i
ui � li � 0, for each i 2 I :
bounds on the amount of input i consumed

Define li � xi � ui, for each i 2 I :
the number of units of input i to be purchased

Minimize
P
i2I cixi

Subject to bo �
P
i2I aoixi � do, for each o 2 O

Figure 3–1. General statement of a minimum-cost input model.

set INPUTS;
set OUTPUTS;

param inLo {INPUTS} >= 0;
param inHi {INPUTS} >= 0;
param inCost {INPUTS} >= 0;

param outLo {OUTPUTS} >= 0;
param outHi {OUTPUTS} >= 0;
param oi {OUTPUTS,INPUTS} >= 0;

var InBuy {i in INPUTS} >= inLo[i], <= inHi[i];

minimize TotalCost: sum {i in INPUTS} inCost[i] * InBuy[i];

subject to OutputLimits {o in OUTPUTS}:
outLo[o] <= sum {i in INPUTS} oi[o,i] * InBuy[i] <= outHi[o];

Figure 3–2. The model of Figure 3–1, rewritten in the AMPL modeling language
(input.mod).

aoi as oi[o,i]. Then the expression for the constraints is transformed into
AMPL by:

bo �
P
i2I aoixi � do -!

outLo[o] <= sum {i in INPUTS} oi[o,i] * InBuy[i] <= outHi[o]

There is one such constraint for each output, so the AMPL model incorporates a
constraint declaration consisting of subject to OutputLimits {o in OUTPUTS}
followed by the above expression.

For the rest of this chapter we’ll use this AMPL model as our prototype of an
input model, deriving from it a series of more specialized models for various
purposes.

3.2 Diet models

The AMPL diet model of Figure 1–4 is readily specialized from the general
input model. The input and output sets are the foods and nutrients:

INPUTS -! FOOD
OUTPUTS -! NUTR

Corresponding to the inputs, the costs and purchase amounts are simply the
same entities with different names, while the limits are trivially zero and infinity:

inCost -! foodCost
inBuy -! Buy
inLo -! 0
inHi -! +Infinity

Corresponding to the nutrients, the lower limits are the same, the upper limits
are infinite, and the input-output coefficients are the amounts of nutrients in
each food:

outLo -! nutrLo
outHi -! +Infinity
oi -! amt

With these interpretations, the input model’s objective becomes that of the diet
model, and the input model’s constraint OutputLimits becomes the constraint
Need of the diet model.

We can make the diet model more general by substituting foodLim and nu-
trHi rather than �1 for inHi and outHi, respectively, as seen in Chapter 2.
Provision could be made for requiring certain foods in the diet by substituting
a general parameter rather than 0 for inLo. Then, as we have also shown in
Chapter 2, the model may be further refined to take account of circumstances
specific to diets or to a particular diet application.

Meal planning represents a further extension, in which foods are assigned
to particular meals (as briefly suggested at the end of Section 2.5) rather than
to the diet as a whole. A meal planning model can be derived from the input
model through replication of variables over meals and addition of constraints

to insure that meals are practical and appetizing. Model extensions based on
replications are the topic of Chapter 6.

For practical purposes, diet models are most likely to be found in institu-
tions — such as schools or hospitals — where close control is more desirable
or necessary than in the fast-food restaurant setting of Chapters 1 and 2. The
most common use is probably for animal diets, but in that case the diet model
begins to resemble the blending model that is our next example.

3.3 Blending models

Closely related to the diet problem is the blending problem in which the
inputs are materials to be physically mixed together rather than merely selected.
Each material contributes characteristic amounts of relevant qualities to the
blend, and the problem is to choose amounts of the materials that will achieve
acceptable amounts of all qualities at minimum cost.

In the conversion from the generic input model of Figure 3–2 to this case,
the sets naturally become

INPUTS -! MATL # materials
OUTPUTS -! QUAL # qualities

The input parameters and variables relate to the amounts to be blended:

inCost -! matCost # material costs per unit
inBuy -! MatBlend # amounts of materials to be blended
inLo -! matLo # lower bounds on amounts blended
inHi -! matHi # upper bounds on amounts blended

The input-output coefficients of the input model become the amounts of each
quality in a unit of each material. The qualities required of the resulting blend
are typically also specified per unit of the blend, however:

outLo -! qualLo # lower bounds on quality per unit of blend
outHi -! qualHi # upper bounds on quality per unit of blend
oi -! amt # quality per unit of each material

Thus while substitution into the objective of the input model is straightforward,
the constraints need to be changed somewhat.

If we make these substitutions in the input model’s expression for total pro-
duction of output, we get an expression for the total of quality q in the blend:

sum {m in MATL} amt[q,m] * MatBlend[m]

Because we are mixing all of the materials together in this case, we can also
speak meaningfully of the total blended material produced. This is simply the
sum of the amounts of the individual materials included in the blend. In the
model, it is the sum of the decision variables:

sum {m in MATL} MatBlend[m]

The quality per unit of the blend is the total quality divided by the total units

set MATL; # materials
set QUAL; # qualities

param matLo {MATL} >= 0; # lower bounds on amounts blended
param matHi {MATL} >= 0; # upper bounds on amounts blended
param matCost {MATL} >= 0; # material costs per unit

param qualLo {QUAL} >= 0; # lower bounds on quality per unit
param qualHi {QUAL} >= 0; # upper bounds on quality per unit
param qual {QUAL,MAT} >= 0; # quality per unit of each material

var MatBlend {m in MATL} >= matLo[m], <= matHi[m];

minimize TotalCost: sum {m in MATL} matCost[m] * MatBlend[m];

subject to OutputLimits {q in QUAL}:
qualLo[q] <=

(sum {m in MATL} qual[q,m] * MatBlend[m])
/ (sum {m in MATL} MatBlend[m]) <= qualHi[q];

Figure 3–3. A blending derivative of the input model of Figure 3–1 (blend1.mod).

— the first expression above divided by the second. It is this quotient that is
required to lie within specified limits:

qualLo[q] <=
(sum {m in MATL} qual[q,m] * MatBlend[m])

/ (sum {m in MATL} MatBlend[m]) <= qualHi[q];

The completed model incorporating these constraints is shown in Figure 3–3.
One of the earliest applications of optimization models was to the refining

of petroleum products such as gasoline. In simplified form, a refinery can be

param: MATL: matLo matHi matCost :=
SRG 7525 21170 9.57 # straight run gasoline
N 100 500 8.87 # naphtha
RF 12250 16140 11.69 # reformate
CG 4100 4610 10.88 # cracked gasoline
B 280 370 6.75 ; # butane

param: QUAL: qualLo qualHi :=
VAP 11.4 12.7 # vapor pressure
OCT 89 91 ; # research octane

param amt: SRG N RF CG B :=
VAP 18.4 6.54 2.57 6.9 119.2
OCT 78.5 65.0 104.0 93.7 91.8 ;

Figure 3–4. Data for the blending model of Figure 3–3, corresponding to a very simple
oil refining example (blend.dat).

regarded as breaking crude oil into simpler collections of intermediate materi-
als, then blending the intermediates into the desired final products. Figure 3–4
shows data for a bit of the blending part of the problem, where five intermedi-
ates are blended to yield premium gasoline having acceptable amounts of two
qualities, vapor pressure and research octane.

We can solve this simple instance using the MINOS solver:

ampl: model blend1.mod;
ampl: data blend.dat;

ampl: option solver minos;

ampl: solve;
MINOS 5.5: optimal solution found.
1 iterations, objective 315923.277
Nonlin evals: constrs = 6, Jac = 5.

ampl: display MatBlend;

MatBlend [*] :=
B 370

CG 4100
N 453.834

RF 12250
SRG 12705.3

The reference to Nonlin evals in the MINOS is a sign that blend.mod is not
linear like our previous examples. We can get AMPL to confirm this by setting
option show stats to 1:

ampl: option show stats 1;

5 variables, all nonlinear
2 constraints, all nonlinear; 10 linear nonzeros
1 linear objective; 5 nonzeros.

Recall that for a constraint to be linear, the expressions on either side of the
comparison operator (= or <= or >=) must be only sums of constant or constant-
times-variable terms. In our constraints we have an expression involving vari-
ables divided by another expression involving variables, which cannot be broken
down into simple linear terms (even though the numerator and denominator are
linear individually).

Because we have formulated a nonlinear model, it can only be handled by
a solver that is designed to deal with nonlinearities. If we try to send it to the
CPLEX solver, for example, it is rejected:

ampl: option solver cplex;
ampl: solve;
CPLEX 8.1.0: /tmp/at7007.nl contains nonlinear constraints.

Although it’s not so clear from this simple example, nonlinear models tend
to be significantly harder to solver than linear ones of comparable size and
structure. Consequently a linear formulation is preferable when one is available.

subject to OutputLo {q in QUAL}:
sum {m in MATL} qual[q,m] * MatBlend[m]

>= qualLo[q] * sum {m in MATL} MatBlend[m];

subject to OutputHi {q in QUAL}:
sum {m in MATL} qual[q,m] * MatBlend[m]

<= qualHi[q] * sum {m in MATL} MatBlend[m];

Figure 3–5. Constraints that should be substituted for OutputLimits (Figure 3–3) to
yield a linear blending model (blend2.mod).

In the case of the blending model, a linear equivalent — one that has the same
solutions but only linear constraints — is readily constructed by multiplying
through by the linear terms in the denominator. We then get:

qualLo[q] * sum {m in MATL} MatBlend[m] # NOT LEGAL SYNTAX
<= (sum {m in MATL} qual[q,m] * MatBlend[m])

<= qualHi[q] * sum {m in MATL} MatBlend[m]

Because there are now variables in all three expressions separated by <= oper-
ators, we need to write this as two separate AMPL constraints, OutputLo and
OutputHi, as shown in Figure 3–5. When these are substituted for the nonlinear
constraint OutputLimits, we can use CPLEX to get the same minimum cost:

ampl: reset;
ampl: model blend2.mod;
ampl: data blend.dat;

ampl: option solver cplex;

ampl: solve;
CPLEX 8.1.0: optimal solution; objective 315923.277
2 dual simplex iterations (0 in phase I)

With the command display MatBlend it is easy to confirm that the linear model
gets the same values for the variables as the nonlinear one.

In the data for this example, the positive lower bounds on the amounts of
input to the blend (parameters matLo[m]) are what force a positive output. If
we set these lower bounds to zero, then it’s optimal to do nothing, at zero cost:

ampl: let m in MATL matLo[m] := 0;
ampl: solve;

CPLEX 8.1.0: optimal solution; objective 0
0 dual simplex iterations (0 in phase I)

ampl: display MatBlend;

MatBlend [*] :=
B 0

CG 0
N 0

RF 0
SRG 0

If the input lower bounds are truly zero, then the model must be lacking another
lower bound of some kind, perhaps on the total weight or volume of the blended
output.

The vapor pressure and research octane in our refinery example are not
actual substances in the blend, and hence the expression sum {m in MATL}
amt[q,m] * MatBlend[m] does not have any physical meaning — it does not
make sense to talk of the “total vapor pressure” in the blend. Instead the quo-
tient in the OutputLimits constraint can be viewed as representing a weighted
average of the qualities in the material components of the blend. This is a
kind of linearity assumption, which justifies our use of the linear expressions in
the numerator and denominator in constraint OutputLimits. Not all physical
substances obey such an assumption; in refining, an early example was the non-
linear effect of a tetraethyl lead component on the octane rating of the blend.

Food processing is another key application area for blending models. Fast
food restaurants of the sort considered in Chapters 1 and 2 use blending mod-
els to determine how they will mix beef from different suppliers to use for ham-
burgers. Processed cheese is another mixed food, but one that involves quite
a few additional complications due to its use of several milk products that can
very considerably by supplier and time of year. Animal feeds are an even larger
and more varied application. As a example, it is desirable to feed hogs with
cheap food processing by-products and wastes as much as possible. But certain
nutritional qualities must be maintained, along with other qualities that ensure
that the hogs can be induced to eat the resulting mix.

3.4 Cutting and packing models

Not all applications of the minimum-cost input model are so immediately ev-
ident as the diet and blending ones. As an example, we next consider a problem
that results when bulk quantities of a resource must be somehow divided up
to meet individual orders. Paper is produced in wide rolls, for example, which
must be slit into narrow ones for individual customers. Steel sheets are pro-
duced in coils for which there is a similar problem. Electrical components must
be placed in racks to use space most economically, or trucks must be loaded in
a way that meets weight requirements efficiently.

As an introduction that’s easy to follow, we’ll start with the cutting of paper
rolls. To make best use of resources, a paper-making machine produces full-
width “raw” rolls that are subsequently cut, on an equally wide slitting machine,
into a variety of narrower “finished” widths. The slitting machine can be set to
produce any pattern of widths, and a particular pattern is distinguished by the
number of rolls of each width that it contains. For example, given master rolls
10000 wide, one pattern consists of two 3600 rolls and two 1400 rolls; another is a
1400 roll, a 3600 roll, a 4500 roll, and 500 of waste.

From the slitting machine’s standpoint, the inputs are the raw rolls that
come off of the paper-making machine. But the decision to be made is how
many finished rolls to cut in each pattern, so that all orders are filled using the
fewest raw rolls — hence leaving the least waste. It follows that, for the input

model, the inputs are patterns and the outputs are finished widths ordered:

INPUTS -! PATTERNS # cutting patterns
OUTPUTS -! WIDTHS # narrow widths ordered by customers

Corresponding to the input patterns we have the following model components:

inCost -! 1
inBuy -! Cut # raw rolls to be cut in each pattern
inLo -! 0
inHi -! +Infinity

The number of raw rolls cut according to pattern p is Cut[p]. The correspond-
ing cost is set to 1 for every pattern, so that the input model’s objective function
reduces to sum {p in PATTERNS} Cut[p] — precisely the total number of raw
rolls cut by all patterns, which is what we seek to minimize.

Corresponding to each output finished width, the quantity ordered provides
a lower limit on the rolls of that width produced by cutting in all patterns.
Since the variety of patterns is limited, however, it may not be possible to meet
all ordered amounts exactly. Contracts thus typically define a small overrun
allowance by which the ordered amount may be exceeded. The allowance can be
expressed as a percentage of the amount ordered, but for this example we take
it to be a fixed amount:

outLo[j] -! order[w] # rolls of width w ordered
outHi[j] -! order[w] + overrun # ... plus allowed overrun

The input-output coefficients give the number of each finished width (output)
from each pattern (input):

oi[i,j] -! rolls[w,p] # rolls of width w in pattern p

set PATTERNS; # patterns
set WIDTHS; # finished widths

param order {WIDTHS} >= 0; # rolls of width j ordered
param overrun; # permitted overrun on any width
param rolls {WIDTHS,PATTERNS} >= 0;

# rolls of width i in pattern j

var Cut {p in PATTERNS} integer >= 0;
# raw rolls to be cut in each pattern

minimize TotalRawRolls: sum {p in PATTERNS} Cut[p];

subject to FinishedRollLimits {w in WIDTHS}:
order[w] <= sum {p in PATTERNS} rolls[w,p] * Cut[p]

<= order[w] + overrun;

Figure 3–6. A roll-cutting derivative of the input model of Figure 3–1 (cut.mod).



A–46 Optimization Models — x3.4

The completed model is shown in Figure 3–6. The input model’s expression for
the amount of output produced from one unit of input has been transformed
to rolls[w,p] * Cut[p], the number of rolls of finished width w produced by
cutting one roll according to pattern p. The sum of this expression over all
patterns p gives the total of finished widths w produced, which is constrained to
lie between the amount ordered and the amount ordered plus the overrun limit.

Figure 3–7 shows a small data instance. For a total of 147 orders ranging
in width from 1400 to 4500, cut from raw rolls of width 10000, there exists a
minimum using only 4 of the patterns:

ampl: model cut.mod;
ampl: data cut.dat;

ampl: option solver cplex;
ampl: solve;

CPLEX 8.1.0: optimal integer solution; objective 45
3 MIP simplex iterations
0 branch-and-bound nodes

ampl: display Cut;

Cut [*] :=
1 0
2 0
3 0
4 20
5 1
6 0
7 20
8 4

Another useful expression to display here is the constraint “body” — the sum
of all the terms involving variables:

set PATTERNS := 1 2 3 4 5 6 7 8 ;

param: WIDTHS: order :=
14 38
31 39
36 61
45 9 ;

param roll: 1 2 3 4 5 6 7 8 :=
14 7 3 2 2 1 0 0 0
31 0 0 2 0 0 3 2 0
36 0 0 0 2 1 0 1 0
45 0 1 0 0 1 0 0 2 ;

param overrun := 3;

Figure 3–7. Data for the model of Figure 3–6 (cut.dat).

ampl: display order, FinishedRollLimits.body;

: order FinishedRollLimits.body :=
14 38 41
31 39 40
36 61 61
45 9 9

We see from this listing that there are overruns of three 1400 rolls and one 3100

roll.
This simple roll-cutting model can be enhanced in various ways for practical

purposes. For example, there is typically a cost in lost time and extra labor
to change the setting of the slitting machine from one pattern to another. To
deal with this situation, a penalty for setups of this kind could be introduced
into the objective, or a pattern could be required to be used on some minimum
number of rolls if it is used at all. Neither of these extensions can be expressed
using only the Cut[j] variables and linear constraints, however; more advanced
approaches are needed, as we’ll see in later discussions of integer programming
formulations.

Another practical challenge is to determine all of the patterns to be con-
sidered. If the ordered widths are numerous and are small relative to the raw
width, then the number of patterns can grow unmanageably large, even if at-
tention is limited to patterns that do not produce much waste. The number of
patterns used in a solution tends to remain small, however, as the sample so-
lution above suggests. Thus various advanced algorithmic schemes have been
devised to combine the generation of patterns with the search for the best cut-
ting solution.

If instead of rolls to be cut into smaller sized orders we imagine bins to be
filled with smaller sized objects, then the entire preceding analysis carries over
to a bin packing problem. The packing may involve physical objects, or it may
be more conceptual in nature, such as in the packing of advertisements into a
column running down a web page, or employees into work shifts. The latter is
sufficiently special to merit its own section below.

We have limited our consideration to “one dimensional” cutting and packing,
which is by far the easiest. Problems in two and three dimensions are frequently
encountered, however, which are more difficult to model as well as to solve.

3.5 Scheduling models

As a final example of minimum-cost input, we consider a problem of work-
force scheduling. At a production or service facility of some kind, we require
a certain number of employees to work in each of a series of time periods, or
shifts. For each week we assign each employee a schedule of shifts to work.
For practical or legal reasons, however, there are only certain schedules that
an employee can be assigned to follow. Thus our problem reduces to deciding
how many employees to assign to each schedule. The objective to be minimized
is the total employees assigned, and the constraints are that the work require-
ments for each shift be met.

How does an input model apply here? The inputs are schedules, and the
outputs are shifts:

INPUTS -! SCHEDS # schedules
OUTPUTS -! SHIFTS # shifts

If this seems odd, you may find it useful to make an analogy with the diet model;
just as each unit of a given food contributes certain amounts of nutrients toward
the diet’s nutritional requirements, each employee working a given schedule
contributes certain shifts of work toward the facility’s labor requirements.

Corresponding to the inputs (schedules) we have the following model com-
ponents:

inCost -! 1
inBuy -! SchedWork # employees to be assigned each schedule
inLo -! 0
inHi -! +Infinity

The number of employees assigned to work schedule j is SchedWork[j]. The
corresponding cost is set to 1 for every schedule, so that the input model’s ob-
jective sum {j in INPUTS} inCost[j] * InBuy[j] reduces to sum {j in SCHEDS}
SchedWork[j] — exactly the total number of employees assigned to work all
schedules, which is what we seek to minimize.

Corresponding to the inputs (shifts), the lower limits are the numbers of
workers required for each shift, and no upper limit is required (at least in this
version):

outLo -! shiftLo # employees required on each shift
outHi -! +Infinity

The input/output coefficients require a little more analysis. In the scheduling
context, oi[i,j] should be the number of units of output (shift) i yielded by
each unit of input (schedule) j. Since both shifts and schedules are measured
in numbers of employees, this becomes the number of employees contributed
to shift i by each employee working schedule j. But each employee working
schedule j contributes either one employee to shift i — if shift i is one of
the shifts in schedule j — or else no employees to shift i — if shift i is not in
schedule j. So the input-output parameters, which we’ll call usedby in this case,
are all either zeros or ones:

oi[i,j] -! usedby[i,j] # 1 if shift i is used by schedule j
# 0 otherwise

The complete model is shown in Figure 3–8.
To produce a specific instance for examination, suppose that we are schedul-

ing for a week, and that there are three shifts a day Monday through Friday, and
two shifts on Saturday. Every schedule is 5 shifts, and may not contain more
than one shift in any 24-hour period, so that for example Monday 1st shift and
Tuesday 2nd shift may be in the same schedule, but Monday 2nd shift and Tues-
day 1st shift may not. There are 126 different schedules that adhere to these
rules. We would like to have at least 100 employees working the first shift each

set SCHEDS; # schedules
set SHIFTS; # shifts

param shiftLo {SHIFTS} >= 0;
# employees required on each shift

param usedby {SHIFTS,SCHEDS} >= 0;
# 1 if shift i used by schedule j, else 0

var SchedWork {j in SCHEDS} integer >= 0;
# employees to be assigned each schedule

minimize TotalEmployees: sum {j in SCHEDS} SchedWork[j];

subject to ShiftLimits {i in SHIFTS}:
sum {j in SCHEDS} usedby[i,j] * SchedWork[j] >= shiftLo[i];

Figure 3–8. A scheduling derivative of the input model of Figure 3–1 (sched.mod).

day, 78 the second shift, and 52 the third shift; we want to determine how many
employees should work each schedule, so that these requirements are met with
the fewest employees possible.

Using the scheduling model of Figure 3–8 for this purpose, we can write the
AMPL data file for this case as shown (in part) in Figure 3–9. It is instructive to

param: SHIFTS: shiftLo :=
Mon1 100 Mon2 78 Mon3 52
Tue1 100 Tue2 78 Tue3 52
Wed1 100 Wed2 78 Wed3 52
Thu1 100 Thu2 78 Thu3 52
Fri1 100 Fri2 78 Fri3 52
Sat1 100 Sat2 78 ;

set SCHEDS :=
M1T1W1T1F1 M1T1W1T1F2 M1T1W1T1F3 M1T1W1T1S1 M1T1W1T1S2
M1T1W1T2F2 M1T1W1T2F3 M1T1W1T2S1 M1T1W1T2S2 M1T1W1T3F3
M1T1W1T3S1 M1T1W1T3S2 M1T1W1F1S1 M1T1W1F1S2 M1T1W1F2S2
...

param usedby (tr): Mon1 Mon2 Mon3 Tue1 Tue2 Tue3 Wed1 Wed2 Wed3
Thu1 Thu2 Thu3 Fri1 Fri2 Fri3 Sat1 Sat2 :=

M1T1W1T1F1 1 0 0 1 0 0 1 0 0 1 0 0 1 0 0 0 0
M1T1W1T1F2 1 0 0 1 0 0 1 0 0 1 0 0 0 1 0 0 0
M1T1W1T1F3 1 0 0 1 0 0 1 0 0 1 0 0 0 0 1 0 0
M1T1W1T1S1 1 0 0 1 0 0 1 0 0 1 0 0 0 0 0 1 0
M1T1W1T1S2 1 0 0 1 0 0 1 0 0 1 0 0 0 0 0 0 1
...

Figure 3–9. Excerpts from data for the model of Figure 3–8, representing a weekly
scheduling example (sched.dat).

first look at the fractional solution returned by the MINOS solver:

ampl: model sched.mod;
ampl: data sched.dat;

ampl: option solver minos;
ampl: solve;

MINOS 5.5: ignoring integrality of 126 variables
MINOS 5.5: optimal solution found.
19 iterations, objective 265.6

ampl: option omit zero rows 1, display 1col 0;
ampl: display SchedWork;

SchedWork [*] :=
M1T1T2F2S2 6.8 M2T2W2T2F2 35.6
M1T1W1T3F3 28.8 M2T2W2T2S1 28
M1T1W2F1S2 6.8 M2T3W3F1S2 -5.87667e-15
M1T1W2T2S1 7.6 M3T3T1F1S2 14.4
M1T1W3F1S2 14.4 M3T3W3F1S1 14.4
M1T2T2F2S2 -4.93961e-15 M3T3W3T3F3 23.2
M1W1T1F1S1 35.6 M3T3W3T3S2 4.68529e-16
M2T2T1F1S1 14.4 T1W1T1F2S2 35.6

The AMPL option display 1col offers, like omit zero rows, a way to influence
the display command’s formatting of results. When set to zero, display 1col
favors a format in which the values are not listed in one single column but are
instead spread out over the page.

It is annoying to have to look past the extremely tiny values, like 4.68529e-16,
that are side-effects of the computer’s representation of numbers to only a finite
accuracy. We can suppress the display of all magnitudes less than some small
value, however, by setting AMPL’s display eps option to that value:

ampl: option display eps .000001;
ampl: display SchedWork;

SchedWork [*] :=
M1T1T2F2S2 6.8 M1W1T1F1S1 35.6 M3T3W3F1S1 14.4
M1T1W1T3F3 28.8 M2T2T1F1S1 14.4 M3T3W3T3F3 23.2
M1T1W2F1S2 6.8 M2T2W2T2F2 35.6 T1W1T1F2S2 35.6
M1T1W2T2S1 7.6 M2T2W2T2S1 28
M1T1W3F1S2 14.4 M3T3T1F1S2 14.4

Because in this case there are no upper bounds on the number of employees
assigned to a shift, we can derive a practical schedule by rounding up each
value to the next highest integer. The resulting objective value is only a few
percent higher:

ampl: print sum {j in SCHEDS} ceil(SchedWork[j]);
273

This seems a bit high for the number of variables being rounded, however, and
this problem is small enough to double-check the result by typing in all the
rounded values:



Draft of September 11, 2005 A–51

ampl: print 7+29+7+8+15+36+15+36+28+15+15+24+36;
271

One source of this discrepancy is already evident from our first display above.
The variable whose value is 4.68529e-16 is being rounded up to 1, when it
should be considered as zero. A further source of trouble can only be seen by
setting the AMPL option display precision to 0, which is interpreted by AMPL
as a request to show all values at maximum precision:

ampl: display precision 0;
ampl: display SchedWork;

SchedWork [*] :=
M1T1T2F2S2 6.799999999999995 M2T2W2T2F2 35.599999999999994
M1T1W1T3F3 28.799999999999997 M2T2W2T2S1 28.000000000000018
M1T1W2F1S2 6.79999999999999 M3T3T1F1S2 14.4
M1T1W2T2S1 7.599999999999998 M3T3W3F1S1 14.4
M1T1W3F1S2 14.40000000000001 M3T3W3T3F3 23.200000000000006
M1W1T1F1S1 35.6 T1W1T1F2S2 35.59999999999999
M2T2T1F1S1 14.399999999999995

The value of 28 seen in the previous listings is here shown to be more accurately
28.000000000000018, as a result of which it is being rounded up to 29. When we
round all values of the variables at, say, the 10th decimal place before rounding
up to the next integer, then we get the result that was expected:

ampl: print sum {j in SCHEDS} ceil(round(SchedWork[j],10));
271

In general, one should never assume that numbers received from a solver are
exactly as shown. Tiny discrepancies may affect comparisons between numbers
as well as roundings.

Switching now to a solver that can enforce integrality constraints, we can
find the minimum number of whole employees needed to meet the shift re-
quirements:

ampl: option solver cplex;
ampl: solve;

CPLEX 8.1.0: optimal integer solution; objective 266
26 MIP simplex iterations
0 branch-and-bound nodes

ampl: display SchedWork;

SchedWork [*] :=
M1T1T1F1S1 6 M1T1W3F1S1 1 M2T2W3T3F3 5 M3W1T1F1S1 19
M1T1W1T1F1 20 M1T2T1F1S1 19 M2T2W3T3S2 4 T1W1T1F2S2 24
M1T1W1T1S2 12 M1W2T2F2S2 17 M2T3W3T3S1 19 T1W2T2F2S2 12
M1T1W1T2F3 24 M2T2W2F1S1 35 M3T3T2F2S2 10
M1T1W1T3S1 1 M2T2W2T2F2 15 M3T3W3T3F3 23

It turns out that we can make do with 266 employees rather than the 271 ob-
tained by rounding up. Since the optimal objective without the integrality con-

straint was 265.6, this is the best we could have hoped for.
Once we have the model and data set up, it is easy to explore alternative

scenarios. As an example, what if we do not want to use schedules that have
more than 5 shifts off between two shifts worked? There are two such schedules
in our solution above: M1T1W1T1S2 has no work between Thursday first shift
and Saturday second shift — see its entry in the last line of Figure 3–9 — and
M1W2T2F2S2 has no work between Monday first shift and Wednesday second
shift. In fact only four of the 126 schedules have this property, so we can use
AMPL’s fix command to force the corresponding variables to zero, after which
we can solve again:

ampl: fix SchedWork["M1W2T2F2S2"] := 0;
ampl: fix SchedWork["M1T1T2F2S2"] := 0;
ampl: fix SchedWork["M1T1W1F2S2"] := 0;
ampl: fix SchedWork["M1T1W1T1S2"] := 0;
ampl: solve;

CPLEX 7.0.0: optimal integer solution; objective 266
21 MIP simplex iterations
0 branch-and-bound nodes
3 simplex iterations (0 in phase I)

ampl: display SchedWork;

SchedWork [*] :=
M1T1W1T1F1 4 M1T1W3T3F3 12 M2T2W2T2F2 32 M3W2T2F2S2 24
M1T1W1T1F3 9 M1T2W3T3F3 12 M2T2W2T3S2 9 T1W1T1F1S1 35
M1T1W1T1S1 4 M1T3T1F1S1 24 M2W1T1F1S1 12
M1T1W1T2S2 22 M2T2T1F1S1 12 M3T3W3F2S2 23
M1T1W1T3F3 14 M2T2W2F1S1 13 M3T3W3T3F3 5

The solution is very different — in fact it has only four schedules in common
with our previous one — but the objective value is unchanged! For this data at
least, it’s “free” to have no more than 5 shifts off between two shifts worked.
Other restrictions can be imposed at only modest cost. As just one example,
if we rule out also the six schedules that have no shift on Monday, then the
optimal solution is again very different but the required number of employees
increases by less than 3.5%.

This example serves as a reminder that solvers typically return an optimal
solution, not the optimal solution. Many alternative solutions may achieve the
same optimum, and a still greater number may be acceptably close to optimal.
Thus the first solution returned by a model such as this is usually only the
beginning of an analysis. Decision-makers often want to have the option of
choosing between several of the solutions generated. This choice might suggest
to the modeler some extra constraints that may be added, but it might reflect
complex or short-term considerations that are not practical to model.

Circumstances may suggest adding constraints to restrict the number of dif-
ferent schedules used, or to rule out the use of a schedule that is worked by
only a few employees. These “logical” constraints require more integer vari-
ables, or else a more general form of constraint than the numerical equalities
and inequalities that we have used so far.

A specialized kind of workforce scheduling appears in airline applications,
where the scheduling applies to whole crews rather than individual employees.
The shifts become flights, and the schedules are series of flights — “pairings”
in airline jargon — that represent one or more whole days of work. A pair-
ing can be assigned at most one crew; each flight must be covered by at least
one, and preferably exactly one, crew. Airlines employ quite sophisticated crew
scheduling models that are based on this simple idea.

4. Output and Input-Output Models

The reverse of a minimum-cost input problem is a maximum-income output
problem. This section gives general and specific examples of models for the
output problem, and then shows how the formulation can be reinterpreted to
encompass input and output within the same model.

4.1 The output optimization problem

Imagine an operation of some sort that can produce a variety of outputs. To
make each unit of a particular output, the operation requires specified amounts
of each of several inputs, with the amount of each input consumed being pro-
portional to the amount of each output produced. Each output also has a value
per unit, again proportional to the amount produced. Our goal is to choose how
much of each output to produce, within specified limits, so that the total value
of the outputs is as great as possible consistent with the amounts of the inputs
available.

We build our mathematical model of this problem, in Figure 4–1, by analogy
with our model for the input problem in Chapter 3. Again the fundamental sets
of entities are inputs I and outputs O. Each unit of an output o 2 O has a value
co, and requires amounts aio of the inputs i 2 I . The units of output o 2 O
produced must lie between lo and uo, and the units of input i 2 I available is
given by bi.

The decision variables are now the amounts of the outputs to be produced,
so we may write them as xo for each output o 2 O. Then since the value of

Given I , a set of inputs

O, a set of outputs

and aio � 0, for each i 2 I and o 2 O:
the units of input i consumed to make one unit of output o
bi � 0, for each i 2 I :
amount of input i available for consumption

co � 0, for each o 2 O:
the value per unit of output o
uo � lo � 0, for each o 2 O:
bounds on the amount of output o produced

Define lo � xo � uo, for each o 2 O:
the number of units of output o to be produced

Maximize
P
o2O coxo

Subject to
P
o2O aioxo � bi, for each i 2 I

Figure 4–1. General statement of a maximum-value output model.

set INPUTS;
set OUTPUTS;

param outLo {OUTPUTS} >= 0;
param outHi {OUTPUTS} >= 0;
param outValue {OUTPUTS} >= 0;

param inHi {INPUTS} >= 0;
param io {INPUTS,OUTPUTS} >= 0;

var OutMake {o in OUTPUTS} >= outLo[o], <= outHi[o];

maximize TotalValue: sum {o in OUTPUTS} outValue[o] * OutMake[o];

subject to InputLimits {i in INPUTS}:
sum {o in OUTPUTS} io[i,o] * OutMake[o] <= inHi[i];

Figure 4–2. The maximum-value output model of Figure 4–1, rewritten in the AMPL
modeling language.

output o produced is proportional to the amount produced, it is given by coxo,
and the objective is to maximize the total of these values,

P
o2O coxo. Similarly

since the amount of input i consumed in making output o is proportional to
the amount of o made, this amount is given by aioxo. The total of input i
consumed by all outputs is thus

P
o2O aioxo, and the constraints say that this

cannot exceed the amount available for consumption,
P
o2O aioxo � bi;

for each input i 2 I .
An analogous description in AMPL is given by Figure 4–2. It resembles Figure

3–2’s input model, with again the sets I and O being written as INPUTS and
OUTPUTS, respectively. The variables xo become OutMake[o], and the limits
lo and uo are outLo[o] and outHi[o]. Thus the conversion of the variable
declaration is

lo � xo � uo, for each o 2 O -!
var OutMake {o in OUTPUTS} >= outLo[o], <= outHi[o]

The values co become outValue[o], so that the expression for the objective is
converted as

P
o2O coxo -! sum {o in OUTPUTS} outValue[o] * OutMake[o]

This expression is named TotalValue and is placed in the maximize declara-
tion.

The available inputs bi are written as inHi[i], and the “input-output co-
efficients” aio may again be denoted io[i,o]. Then the expression for the
constraints is transformed into AMPL by:

P
o2O aioxo � bi -!
sum {o in OUTPUTS} io[i,o] * OutMake[o] <= inHi[i]

There is one such constraint for each output, so Figure 4–2’s constraint declara-
tion consists of subject to InputLimits {i in INPUTS} followed by the above
expression.

4.2 A product mix model

Product mix is a typical application of output optimization. Given a range of
products that can be made from available raw materials, the goal of a product
mix model is to determine how much of each product to make so that the total
value of production is as great as possible.

The products are our outputs in this situation, and the output-related com-
ponents of the Figure 4–2 model specialize accordingly:

OUTPUTS -! PROD # products

outLo -! prodLo # lower bounds on units produced
outHi -! prodHi # upper bounds on units produced

outValue -! prodVal # products’ values per unit
OutMake -! ProdMake # units of each product to be made

The raw materials are our inputs, so we also have the following straightforward
set and parameter correspondences:

INPUTS -! RAW # stages of processing

inHi -! rawHi # amounts of raw material available

io[i,o] -! raw2p[r,p] # units of raw material
# to make each product

With these substitutions made in the objective and constraints, we have the
elementary product mix model shown in Figure 4–3.

set RAW; # raw materials
set PROD; # products

param prodLo {PROD} >= 0; # lower bounds on products made
param prodHi {PROD} >= 0; # upper bounds on products made
param prodVal {PROD} >= 0; # values per ton of products made

param rawHi {RAW} >= 0; # raw material amounts available
param raw2p {RAW,PROD} >= 0;

# units of raw material to make a unit of product

var ProdMake {p in PROD} >= prodLo[p], <= prodHi[p];
# units of each product to be made

maximize TotalValue:
sum {p in PROD} prodVal[p] * ProdMake[p];

subject to RawLimits {r in RAW}:
sum {p in PROD} raw2p[r,p] * ProdMake[p] <= rawHi[r];

Figure 4–3. A product mix model derived from Figure 4–2’s maximum-value output
model (prodmix.mod).

param: PROD: prodLo prodHi prodVal :=
2716 100 600 250
2717 50 400 300
2903 75 350 290 ;

param: RAW: rawHi :=
drpu 150000
flex 100000 ;

param raw2p: 2716 2717 2903 :=
drpu 200 200 200
flex 200 140 160 ;

Figure 4–4. A sample of data for Figure 4–3’s product mix model (prodmix.dat).

Using the small example of data shown in AMPL format in Figure 4–4, we get
the following solution:

ampl: model prodmix.mod;
ampl: data prodmix.dat;

ampl: solve;
MINOS 5.5: optimal solution found.
2 iterations, objective 188500

ampl: display prodLo,ProdMake,prodHi;

: prodLo ProdMake prodHi :=
2716 100 100 600
2717 50 400 400
2903 75 150 350

The solution is seen to produce as much as possible of product 2717, and
as little as possible of 2716; the amount of 2903 produced is between its
production bounds. The consumption of raw materials can also be summarized,
as follows:

ampl: display RawLimits.body, RawLimits.slack;

: RawLimits.body RawLimits.slack :=
drpu 130000 20000
flex 1e+05 1.45519e-11 ;

AMPL interprets RawLimits.body as the sum of all terms involving variables
in the constraints RawLimits, and RawLimits.slack as the difference between
the two sides of the RawLimits inequality. The slack of essentially zero for
flex indicates that all of the stocks of that material are used.

4.3 A production planning model

The inputs to a maximum-value output model need not be physical quan-
tities of anything. As a second example we consider a simple factory that
can be viewed, particularly for long-term planning purposes, as converting raw



Draft of September 11, 2005 A–59

“blanks” to various semi-finished intermediate forms. The problem is to choose
the most profitable mix of intermediate forms to make in the next planning
period, subject to the limited availability of production facilities.

The intermediate forms are clearly the outputs in this case. There may be
positive lower bounds and finite upper bounds on production of the intermedi-
ates, depending on known orders and likely demands. Hence we specialize the
output part of Figure 4–2 as follows:

OUTPUTS -! FORMS # semi-finished intermediate forms

outLo -! formLo # lower bounds on units produced
outHi -! formHi # upper bounds on units produced

outValue -! formValue # forms’ values per unit
OutMake -! FormMake # units of each form to be produced

All of the outputs from this facility are made from the same blanks, which we
may imagine for now are available in sufficient quantity to produce whatever
quantities of intermediate forms we want. The limited input is rather the time
available to carry out the stages of the processing:

INPUTS -! STAGES # stages of processing
inHi -! stageHi # hours available at stages

The input-output coefficients would literally be hours (of input) per unit (of
output), but this is typically a very small number. People more naturally think
of production in terms of units per hour, so we write it that way in the model:

io[i,o] -! 1/rate[s,f] # hours per unit of intermediate form
# produced at each stage

set STAGES; # stages of production
set FORMS; # semi-finished intermediate forms

param formLo {FORMS} >= 0; # lower bounds on forms made
param formHi {FORMS} >= 0; # upper bounds on forms made
param formValue {FORMS} >= 0; # values per ton of forms made

param stageHi {STAGES} >= 0; # hours available at stages
param rate {STAGES,FORMS} >= 0;

# units per hour of each form at each stage

var FormMake {f in FORMS} >= formLo[f], <= formHi[f];
# units of each form to be produced

maximize TotalValue:
sum {f in FORMS} formValue[f] * FormMake[f];

subject to InputLimits {s in STAGES}:
sum {f in FORMS} (1/rate[s,f]) * FormMake[f] <= stageHi[s];

Figure 4–5. A production derivative of the output model of Figure 4–2, with the inputs
being available production time in successive stages (prodplan.mod).

param: STAGES: stageHi :=
shape 35
drill 40 ;

param: FORMS: formLo formHi formValue :=
side 1000 6000 25
top 500 4000 30
bottom 750 3500 29 ;

param rate (tr): shape drill :=
side 200 200
top 200 140
bottom 200 160 ;

Figure 4–6. Data for the production output model of Figure 4–5, corresponding to a
very simple steel processing example (prodplan.dat).

The finished model incorporating these substitutions is shown in Figure 4–5.
Figure 4–6 is a small corresponding data instance. Its solution is found as

follows:

ampl: model prodplan.mod;
ampl: data prodplan.dat;

ampl: solve;
MINOS 5.5: optimal solution found.
4 iterations, objective 190071.4286

ampl: display FormMake;

FormMake [*] :=
bottom 3142.86

side 3357.14
top 500

We need not be concerned about an integer solution here, as fractions of one
unit are negligible when production quantities are in the hundreds and thou-
sands. Indeed, for a long-term planning model of this sort, results to 6 digits
— like the 3142.86 and 3357.14 units of intermediate forms bottom and side
above — are meaninglessly precise. They can actually make the results appear
less realistic to the planners who will look at the results. Thus we might want to
instruct AMPL to display only 2 digits of precision:

ampl: option display precision 2;
ampl: display FormMake;

FormMake [*] :=
bottom 3100

side 3400
top 500

In this case the same results would be obtained by instead setting AMPL option
display round to �2, which specifies rounding to two places left of the decimal
point.

4.4 The input-output optimization problem

Whether the decision variables represent amounts of input as in Chapter 3,
or output as in the present chapter, our models are limited to the simplest
kinds of production environments. Many applications involve a more general
collection of operations, with the outputs of one possibly serving as the inputs
of others. Almost the same form of model serves to handle this generalization,
however, provided the data and variables are interpreted in a broader way.

We begin by reinterpreting the set O from the output model (Figure 4–1) as
containing not merely outputs, but activities that may involve a variety of in-
puts and outputs (Figure 4–7). Likewise we reinterpret the set I as containing
not merely inputs, but materials that may be inputs or outputs of various activ-
ities. The variables xo are now the levels at which the activities are operated,
measured in appropriate units.

The values aio are now true input-output coefficients of the model. If aio > 0
then, by analogy with the output model, it can be taken to represent the units
of material i consumed as input to one unit of activity j. If aio < 0, however,
it has the opposite interpretation, as the units of material i produced as output
from one unit of activity j. (If aio � 0 then material i does not figure at all in
activity j.)

Given this expanded definition, the expression aioxo represents either the
consumption of material i due to activity o (if aio > 0) or minus the produc-

Given I , a set of materials

O, a set of activities

and aio, for each i 2 I and o 2 O:
if > 0, units of material i consumed by one unit of activity o
if < 0, units of material i produced by one unit of activity o

bi, for each i 2 I :
if > 0, bound on the amount of material i consumed
if < 0, bound on the amount of material i produced

co, for each o 2 O:
if > 0, the net income per unit of activity o
if < 0, the net expense per unit of activity o

uo � lo � 0, for each o 2 O:
bounds on the level of activity o

Define lo � xo � uo, for each o 2 O:
the level at which activity o is to be run

Maximize
P
o2O coxo

Subject to
P
o2O aioxo � bi, for each i 2 I

Figure 4–7. General statement of a maximum-income input-output model.

tion of material i due to activity o (if aio < 0). Summing over all activities,P
o2O aioxo represents the total consumption of iminus the total production of

i; we’ll refer to this as the net consumption of material i.
Now we can turn to the constraint formulations. Some materials may be

treated, as in our output model, as scarce resources: the amount used must be
at most a certain limit. Then we would like the net consumption to be at most
the maximum available. For this case we can let bi > 0 represent the amount of
material i available, and the constraint is

Pn
j�1 aioxo � bi:

Other materials are intended as products: the amount produced must be at least
a certain requirement. Here we would like the net production — which is simply
the negative of the net consumption — to be at least a certain limit. For this
case we can let bi < 0 represent minus the amount of material i required, and
the constraint is

�Pnj�1 aioxo � �bi:
Multiplying through by �1, this is just the same constraint as the previous one.
Our conventions about the signs of the aio and bi, summarized in Figure 4–7,
allow the same algebraic form to describe both the consumption upper limits
and the production lower limits.

It remains to describe the objective. Again, we use a sign convention to
accommodate two possibilities: when co > 0, it is the net income per unit of
activity o; when co < 0, it is minus the expense per unit of activity o. Then
coxo represents either the net income generated by activity o, or minus the
net expense due to activity o. Summing over all activities,

Pn
j�1 coxo is the

total income minus total expense over all activities, or in other words the total
net income for the entire operation. Naturally, this is a quantity we would be
interested in maximizing.

The formulas for the objective and constraints in the input-output model
(Figure 4–7) are now seen to be exactly the same as for the simpler output model
(Figure 4–1). All that we have done is to interpret the signs of the parameters
more generally, rather than requiring them to be nonnegative in all cases. (As
you might expect, the input model of Figure 3–1 can similarly be turned into a
minimum-cost input-output model, by adoption of an opposite sign convention.)

The corresponding AMPL model, shown in Figure 4–8, naturally looks much
like the one for the output model in Figure 4–2. We have only changed some of
the names and dropped the nonnegativity restrictions on the parameters where
appropriate.

4.5 A multi-facility production model

As an example of the input-output model, we consider a simplified applica-
tion to steel production. At the “hot” end of the process, various components
are mixed in a blast furnace and fed to a continuous caster to produce slabs of
steel. At the (relatively) “cold” end, the slabs are processed into bands, plates,
and coils. Slabs are an entirely intermediate material, being produced at the hot

end for consumption at the cold end. Less obviously, steel scrap produced as a
by-product at the cold end is recycled as an input to the hot end.

We’ll use the AMPL model of Figure 4–8 directly for this example, with the
data shown in Figure 4–9. There are four activities, castSlabs at the hot end
and three at the cold end, and there are 11 materials that figure in one or more
activities.

Recall that positive values in the io table represent amounts of materials
consumed, while negative values represent amounts produced. Thus the first
column in io shows that one unit of castSlabs consumes 0.224 tons of ore,
0.5316 tons of coke, and so forth for limestone, pellets, sinter, and scrap,
to produce 1 ton of slabs.

The remaining activities use a material rolltime that requires a little further
explanation. There is just one rolling mill at the cold end that produces all of
the end products. Thus the activities rollBands, rollCoils, and rollPlate all
compete for the time available at that rolling mill. To incorporate this situation
into the model, we require the three cold-end activities to consume rolltime,
of which only 40 hours are available as you can see from the table of material
values in Figure 4–9.

Looking now at the rollBands column in the io table, we see that a ton of
this activity consumes 1.024 tons of slabs and .000714 hours of rolling time (the
positive entries), to produce one ton of bands and .0115 tons of recyclable scrap
(the negative entries). The entries for rollCoils and rollPlate are interpreted
similarly.

Following the sign conventions of the input-output model, positive values
in the materials data in Figure 4–9 represent bounds on amounts available for
consumption, while negative values represent bounds on amounts required of
production. Thus in particular the negative entries for bands, coils, and plate
indicate requirements of 10000, 20000, and 30000 tons, respectively. For slabs

set MATS; # materials
set ACTS; # activities

param actLo {ACTS} >= 0;
param actHi {ACTS} >= 0;
param actValue {ACTS};

param matHi {MATS};
param io {MATS,ACTS};

var ActLevel {a in ACTS} >= actLo[a], <= actHi[a];

maximize TotalValue: sum {a in ACTS} actValue[a] * ActLevel[a];

subject to MatLimits {m in MATS}:
sum {a in ACTS} io[m,a] * ActLevel[a] <= matHi[m];

Figure 4–8. The maximum-income input-output model of Figure 4–7, rewritten in the
AMPL modeling language (inoutput.mod).

param: MATS: matHi :=
ore Infinity
coke Infinity
limestone Infinity
pellets Infinity
sinter Infinity
slabs 0
bands -10000
coils -20000
plate -30000
scrap Infinity
rolltime 40 ;

param: ACTS: actLo actHi actValue :=
castSlabs 0 Infinity -6.17
rollBands 0 Infinity 6.99
rollCoils 0 Infinity 18.86
rollPlate 0 Infinity 23.15 ;

param io: castSlabs rollBands rollCoils rollPlate :=
ore 0.224 0 0 0
coke 0.5316 0 0 0
limestone 0.1346 0 0 0
pellets 1.0619 0 0 0
sinter 0.2396 0 0 0
slabs -1 1.024 1.0199 1.0199
bands 0 -1 0 0
coils 0 0 -1 0
plate 0 0 0 -1
scrap 0.0181 -.0115 -.0099 -.0160
rolltime 0 .000714 .000478 .000478 ;

Figure 4–9. Data for the input-output model of Figure 4–8, corresponding to a very
simple steel processing example (steel.dat).

the bound entry is zero, reflecting the fact that slabs are only an intermediate
of the steelmaking process; in this case, at least, their availability comes only
through production at the hot end, and their use is only through consumption
at the cold end.

There is also a sign convention for actValue, which tells us that casting
slabs costs 6.17 per ton, while the rolling activities produce outputs at various
net profits per ton. The maximal net profit with this data is found to be as
follows:

ampl: model inoutput.mod;
ampl: data steel.dat;

ampl: solve;
MINOS 5.5: optimal solution found.
1 iterations, objective 1079764.703

As might be expected, only the most profitable product, plate, is rolled beyond
the minimum required:

ampl: display {a in ACTS}
ampl? (ActLevel[a], io[’scrap’,a] * ActLevel[a]);

: ActLevel[a] io[’scrap’,a]*ActLevel[a] :=
castSlabs 80352.8 1454.39
rollBands 10000 -115
rollCoils 20000 -198
rollPlate 48744.8 -779.916

We show also above the amount of scrap used or produced by each activity. It
is interesting to look at the sum of these values, which is the net total scrap:

ampl: display sum {a in ACTS} io["scrap",a] * ActLevel[a];
sum{a in ACTS} io[’scrap’,a]*ActLevel[a] = 361.469

The net total comes out positive, reflecting a need to use more scrap input to
the blast furnace than just the scrap available from the rolling mills.

5. Transportation Models

Together with input and output, “network flow” is the most common pri-
mary source of linear programs. This chapter focuses on the simplest network
models, those known as transportation models. More general network models
will be the subject of later chapters.

After exhibiting in Section 5.1 a small problem of transporting ore from
mines to mills, we show in Section 5.2 how the underlying idea can be abstracted
as a general mathematical transportation model. We can then readily write the
model in AMPL and solve several problem instances, as shown in Section 5.3.

A transportation problem need not involve “transporting” things in a literal
sense, moreover. In Section 5.4 we give an example of a transportation model
adapted to “assigning” students to project groups. Section 5.5 concludes this
chapter with examples of transportation and assignment models generalized to
address common practical concerns.

5.1 A small transportation problem

A steel company mines iron ore at two locations, and must ship the ore to
four mills, as shown in this network diagram:

1

2

3

4

1

2

230

150

80

100

60

130

For a certain production period, the tons of ore available at each mine are as
given at the left of the diagram:

Mine Tons available

1 230
2 150

The tons of ore required at each mill are as given at the right:

Mill Tons needed

1 80
2 100
3 60
4 130



A–68 Optimization Models — x5.1

Finally, there is a table of costs per ton for shipping ore from mines to mills:

Cost to:
1 2 3 4

Cost from: 1 65 40 30 15
2 10 35 40 60

The transportation problem is to decide how much ore to ship from each mine
to each mill, so as to keep the total shipping cost as low as possible while re-
specting all availabilities and meeting all requirements.

To formulate this problem as a linear program, we first define decision vari-
ables that represent the amounts shipped:

Tons to:
1 2 3 4

Tons from: 1 x11 x12 x13 x14

2 x21 x22 x23 x24

The cost of all shipments from mine 1 to mill 1 is the cost per ton (from the
above table) times the number of tons, or 65x11. The cost of all shipments from
mine 1 to mill 2 is similarly 40x12, and so forth. Adding all such terms together,
the total cost of all shipments is

65x11 � 40x12 � 30x13 � 15x14 � 10x21 � 35x22 � 40x23 � 60x24:

This is the objective, to be minimized, of the transportation problem.
Now to the availability constraints. The amount of ore shipped out of mine 1

is the total of the amounts shipped from that mine to all four mills: x11�x12�
x13 � x14. Since only 230 tons are available at mine 1, we have the following
constraint:

x11 � x12 � x13 � x14 � 230:

At mine 2 there is an analogous constraint, that x21 � x22 � x23 � x24 � 150.
Similar reasoning applies to the requirement constraints. The amount of ore

shipped into mill 1 is the sum of the amounts shipped from both the mines:
x11 �x21. Since 80 tons are required, we must impose the following constraint:

x11 � x21 � 80:

We have analogous constraints that x12 � x22 � 100 at mill 2, x13 � x23 � 60 at
mill 3, and x14 � x24 � 130 at mill 4.

Finally, the amounts of the shipments must be nonnegative: x11 � 0, x12 �
0, and so forth. Putting all the constraints together, we have the following linear
program:

Minimize 65x11 � 40x12 � 30x13 � 15x14�
10x21 � 35x22 � 40x23 � 60x24

Subject to x11 � x12 � x13 � x14 � 230
x21 � x22 � x23 � x24 � 150

x11 � x21 � 80
x12 � x22 � 100
x13 � x23 � 60
x14 � x24 � 130

x11 � 0; x12 � 0; x13 � 0; x14 � 0
x21 � 0; x22 � 0; x23 � 0; x24 � 0

The objective is linear because we have assumed that shipment costs are propor-
tional to the amounts shipped. The constraints are automatically linear because
they involve just simple sums of the variables. Notice that each variable appears
in exactly one of the two mine constraints, and in exactly one of the four mill
constraints.

The total ore available in this problem is 380 tons, while the total required
is 370. It’s easy to see that the minimum-cost solution will ship only 370 tons,
exactly meeting all requirements, while 10 tons will be left over at one of the
mines. In any solution that ships any part of the extra 10 tons, the total cost
can only be higher. Thus the mill constraints could be made equalities rather
than inequalities. If the tons available were to equal the tons required, then all
available ore would have to be shipped and all the constraints could be made
into equalities.

Often this kind of model is extended by imposing upper bounds on the flows.
For example, if there is capacity to ship at most 50 tons from mine 1 to mill 4,
then we can add the x14 � 50. Like all simple bounds, this is trivially a linear
constraint.

5.2 A general transportation model

In general terms, we can think of the transportation model as involving ship-
ments of some material from a set O of origins to a set D of destinations. A
complete formulation is shown in Figure 5–1, where material amounts ao are
available at origins o and amounts bd are required at destinations d.

The decision variables are the amounts xod to be shipped from origins o
to destinations d. The amounts shipped must lie between lower limits lod and
upper limits uod. Thus we have the bounds lod � xod � uod.

Shipments from origin o to destination d have cost cod per unit. The cost
of the shipments from o to d is the cost per unit times the number of units
shipped, or codxod, and the total cost of all shipments — the objective to
be minimized — is the sum of these costs over all origin-destination pairs:P
o2O

P
d2D codxod.

At each fixed origin o, the amounts shipped out are xod for every different
destination d. Thus the total shipped out is the sum of these shipments over
all destinations, or

P
d2D xod. This quantity must be at least some amount

Given O, a set of origins

D, a set of destinations

and 0 � alo � aho , for each o 2 O:
bounds on units of material shipped from origin o
0 � bld � bhd , for each for each d 2 D:
bounds on units of material received at destination d
cod � 0, for each o 2 O and d 2 D:
cost per unit to ship material from origin o to destination d

0 � lod � uod, for each o 2 O and d 2 D:
bounds on the units of material shipped
from origin o to destination d

Define lod � xod � uod, for each o 2 O and d 2 D:
units of material that are to be shipped
from origin o to destination d

Minimize
P
o2O

P
d2D codxod

Subject to alo �
P
d2D xod � aho , for each o 2 O

bhd �
P
o2O xod � bld, for each d 2 D

Figure 5–1. A general algebraic statement of transportation model.

alo (which may be zero) and must not exceed the amount available, aho , so the
constraint at origin o is

alo �
P
d2D xod � ah:

There must be one of these constraints for each origin o 2 O.
At each fixed destination d, the amounts shipped in are xod for every dif-

ferent origin o. Thus the total shipped in is the sum of these shipments over
all origins, or

P
o2O xod. This quantity must be at most some amount bhd (which

may be infinity), must not be less than the amount required, bld, so the con-
straint at destination d is

bhd �
P
o2O xod � bld:

There must be one of these constraints for each destination d 2 D.
Although “xod” appears in every summation of the Figure 5–1 formulation, it

is summed differently in each case. In the objective, the summation
P
o2O

P
d2D

is over both origins and destinations, in order to give the total cost over the
shipments from all origins to all destinations. In the first constraint line, the
summation

P
d2D is over the destinations only. This signifies that for each

different origin o, we have the sum over all destinations, giving the total amount
shipped out of o. The second constraint line is the reverse: the summation

P
o2O

is over the origins only, to signify that for each different destination d we have
the sum over all origins, giving the total amount shipped into d.

5.3 Solving transportation model instances

To solve specific instances of the transportation model, we follow the pro-
cedure established in Chapter 3 by first transcribing the mathematical model
(Figure 5–1) to the AMPL modeling language (Figure 5–2). Here we have taken
care to make the AMPL description as much like the mathematical one as possi-
ble, so it is easy to see the correspondences.

Again as in Chapter 3, however, we find it convenient when working with
AMPL to replace the single letters typical of the mathematical form with more
easily remembered names for model components. This consideration leads us
to convert the AMPL model in Figure 5–2 to the one in Figure 5–3. In the de-
scription of the data, sets O and D become ORIG and DEST, parameters al and
ah become origLo and origHi, and so forth. The variables x are renamed
Ship. The objective is called TotalCost, and the two collections of constraints
become OrigAvl and DestReq.

Figure 5–4 shows how the data tables for Section 5.1’s small transporta-
tion problem are transcribed to an AMPL data file. The numbers that we previ-
ously used to identify origins and destinations are replaced by more meanngful
names. After solving the transportation model instance specified by these ta-
bles, we can use a few AMPL display options to produce a table of shipment
amounts in the same layout as the Section 5.1 tables:

set O; # origins
set D; # destinations

param al {O} >= 0; # limits on total units
param ah {o in O} >= al[o]; # shipped from each origin

param bl {D} >= 0; # limits on total units
param bh {d in D} >= bl[d]; # received at each destination

param c {O,D} >= 0; # shipping costs per unit

param l {O,D} >= 0;
param u {o in O, d in D} >= l[o,d];

# individual limits on units shipped

var x {o in O, d in D} >= l[o,d], <= u[o,d];
# individual amounts to be shipped

minimize Z: sum {o in O} sum {d in D} c[o,d] * x[o,d];

subject to OC {o in O}: al[o] <= sum {d in D} x[o,d] <= ah[o];
subject to DC {d in D}: bh[d] >= sum {o in O} x[o,d] >= bl[d];

Figure 5–2. The model of Figure 5.2, transcribed as directly as possible to the AMPL
modeling language.

set ORIG; # origins
set DEST; # destinations

param origLo {ORIG} >= 0; # limits on total units
param origHi {o in ORIG} >= origLo[o]; # shipped from origins

param destLo {DEST} >= 0; # limits on total units
param destHi {d in DEST} >= destLo[d]; # shipped to destinations

param shipCost {ORIG,DEST} >= 0; # shipping costs per unit

param shipLo {ORIG,DEST} >= 0;
param shipHi {o in ORIG, d in DEST} >= shipLo[o,d];

# individual limits on shipments

var Ship {o in ORIG, d in DEST} >= shipLo[o,d], <= shipHi[o,d];
# individual amounts to be shipped

minimize TotalCost:
sum {o in ORIG} sum {d in DEST} shipCost[o,d] * Ship[o,d];

subject to OrigAvl {o in ORIG}:
origLo[o] <= sum {d in DEST} Ship[o,d] <= origHi[o];

subject to DestReq {d in DEST}:
destHi[d] >= sum {o in ORIG} Ship[o,d] >= destLo[d];

Figure 5–3. A more readable version of the AMPL model of Figure 5–2, using longer
names that are easier to understand and remember (transp.mod).

param: ORIG: origLo origHi :=
Chisholm 0 230
Hibbing 0 150 ;

param: DEST: destLo destHi :=
BHarb 80 Infinity
Cleve 100 Infinity
EChic 60 Infinity
Gary 130 Infinity ;

param shipCost: Gary EChic BHarb Cleve :=
Hibbing 65 40 30 15
Chisholm 10 35 40 60 ;

param shipLo default 0 ;
param shipHi default Infinity ;

Figure 5–4. Data tables for using the AMPL model of Figure 5–3 (transp.dat) to solve
Section 5.1’s example, with the cities named rather than numbered. All entries for
shipLo[o,d] and shipHi[o,d] are set by default to 0 and 1, respectively.

ampl: model transp.mod;
ampl: data transp.dat;

ampl: solve;
MINOS 5.5: optimal solution found.
4 iterations, objective 7600

ampl: option display 1col 0, display transpose -10;

ampl: display Ship;

Ship [*,*]
: BHarb Cleve EChic Gary :=
Chisholm 30 0 60 130
Hibbing 50 100 0 0

Exactly 5 � jOj � jDj � 1 origin-destination pairs are used in the solution, and
in fact it can be shown that there is always an optimal solution having at most
this many variables away from their bounds.

The “material” that is “shipped” in a transportation problem need not be
a commodity such as the iron ore in our example. In other applications it is
people being transported, empty boxcars being reallocated, or telephone calls
being routed. As an example, Figure 5–5 shows data for shipments of truck-
loads of goods from 3 origins to 7 destinations. In this situation we want all
of the shipments to be in integer amounts, as it is uneconomical to dispatch a
partially loaded truck, but in fact the solution comes out integral even without
the variables Ship being declared as integer in the AMPL model:

param: ORIG: origLo origHi :=
CLEV 0 15 GARY 0 30 PITT 0 30 ;

param: DEST: destLo :=
DET 9 FRA 12 FRE 6 LAF 4
LAN 17 STL 11 WIN 10 ;

param destHi default Infinity;

param shipCost:
DET FRA FRE LAF LAN STL WIN :=

CLEV 39 14 11 14 16 82 8
GARY 27 9 12 9 26 95 17
PITT 24 14 17 13 28 99 20 ;

param shipLo default 0 ;

param shipHi:
DET FRA FRE LAF LAN STL WIN :=

CLEV 7 7 7 7 7 7 5
GARY 12 12 20 12 20 20 20
PITT 12 12 20 20 20 20 20 ;

Figure 5–5. Data tables for another instance of the Figure 5–3 model (tranint.dat),
where shipments are measured in truckloads and an integer solution is desired.

ampl: reset data;
ampl: data tranint.dat;

ampl: solve;
MINOS 5.5: optimal solution found.
16 iterations, objective 1948

ampl: display Ship;

Ship [*,*]
: DET FRA FRE LAF LAN STL WIN :=
CLEV 0 0 0 0 4 7 4
GARY 0 12 6 4 0 4 6
PITT 9 0 0 0 13 0 0

In fact it can be shown that transportation problems have a strong integrality
property: if all the availabilities, requirements, and bounds are integral, then
(if there is any optimal solution at all) there is an optimal solution that is in-
tegral. The solver that we’re using here is moreover an implementation of a
computational method, the simplex method, which is guaranteed to find such a
solution.

Transportation model instances can easily become very large; even with only,
say, 20 origins and 40 destinations, there are 20 � 40 � 800 variables. At most
20�40�1 � 59 variables need be away from their bounds in an optimal solution,
however, and if the lower bounds are zero then the great majority of optimal
shipment amounts will usually also be zero. In fact transportation problems are
significantly easier to solve than many other optimization problems of compa-
rable size.

5.4 Assignment models

In another kind of transportation problem, items at the origins are merely
associated with destinations, rather than being “transported” in any way. The
result is an assignment problem whose decision variables represent numbers
of items of various types assigned to various recipients or openings.

The applications of assignment models span a great variety of situations.
Here we’ll introduce them through one example, involving the assignment of
students to project groups. We will build an AMPL model for this case by spe-
cializing the AMPL transportation model of Figure 5–3.

We consider the situation that arises in an engineering design course that
sends teams of 3 or 4 students to work on projects at local companies. To en-
sure a good fit, the students are asked to express their preferences for projects
by assigning a rank of 1 to their top choice, 2 to their next choice, and so forth.
We evaluate an assignment by summing the rankings of the students for the
projects to which they are assigned. An assignment that minimizes this sum
can be considered to be “most preferred” overall.

A transportation model specialized for this purpose is shown in Figure 5–6.
Since we have no reason to assume that any two students have the same pref-
erences, we treat each student as a separate “origin” in the model, while the
destinations are the projects to which the students are to be assigned:

set STU; # students
set PRJ; # projects

param teamSizeLo {PRJ} integer >= 0;
param teamSizeHi {p in PRJ} integer >= teamSizeLo[p];

# limits on project team sizes

param assignRank {STU,PRJ} integer >= 1, <= card {PRJ};
# student rankings for projects, 1 for the top ranked

var Assign {STU,PRJ} >= 0, <= 1;
# Assign[s,p] is 1 if student s is assigned project p
# Assign[s,p] is 0 otherwise

minimize TotalRank:
sum {s in STU, p in PRJ} assignRank[s,p] * Assign[s,p];

subject to StuAssign {s in STU}:
sum {p in PRJ} Assign[s,p] = 1;

subject to PrjAssign {p in PRJ}:
teamSizeLo[p] <= sum {s in STU} Assign[s,p] <= teamSizeHi[p];

Figure 5–6. An assignment model for grouping students into project teams so as to
maximize the sum of students’ preferences for the projects to which they are assigned
(assign.mod).

ORIG -! STU # students
DEST -! PRJ # projects

Every student must be assigned. Viewed as a separate “origin,” the lower and
upper number of assignments for each student are both 1. For the projects, the
lower and upper number of assignments are the limits on the project team size.
Thus we have:

origLo -! 1 # each student assigned to
origHi -! 1 # exactly one team

destLo -! teamSizeLo # lower and upper limits
destHi -! teamSizeHi # on size of teams

Because origLo and origHi both become 1, the inequality constraint OrigAvl
in Figure 5–3 becomes 1 <= sum {d in DEST} Ship[o,d] <= 1, which collapses
to an equality.

It remains to reinterpret as assignments the transportation model’s refer-
ences to shipments. We would like to claim the following:

shipLo -! 0
shipHi -! 1

shipCost -! assignRank

Ship -! Assign # Assign[s,p] = 1 if student s put in group p
# Assign[s,p] = 0 if not

set STU :=
Cane Gentil Chung Michael Goldrich Matthew Grasparil Michelle
Henry Steve Jurishica Carley Laurie Alina Lawless Katie
Morrow Timothy Rincones Daianna Shakoor Kashif Sheth Vihar
Stewart Eric Urquhart Geoff Walton Caroline Welsch Todd
Wible Barry Yee Christine Yuen Gigi ;

param: PRJ: teamSizeLo teamSizeHi :=
Abbott 3 4
GM EMD 3 4
Littelfuse 3 4
NMFF 3 4
UOP 3 4
Zebra 3 4 ;

param assignRank: Abbott GM EMD Littelfuse NMFF UOP Zebra :=
Cane Gentil 5 4 3 1 2 6
Chung Michael 1 4 2 5 2 6
Goldrich Matthew 2 5 6 1 3 3
Grasparil Michelle 3 6 4 1 2 5
Henry Steve 5 1 4 6 3 2
Jurishica Carley 2 3 5 1 4 6
Laurie Alina 4 5 2 1 2 6
Lawless Katie 6 1 5 3 2 4
Morrow Timothy 6 5 4 3 2 1
Rincones Daianna 1 2 3 5 4 6
Shakoor Kashif 2 3 1 6 4 5
Sheth Vihar 1 2 3 5 4 6
Stewart Eric 4 2 6 1 4 3
Urquhart Geoff 5 1 4 6 3 2
Walton Caroline 5 1 3 6 4 2
Welsch Todd 5 1 2 6 4 2
Wible Barry 6 3 1 5 3 2
Yee Christine 4 5 2 1 2 6
Yuen Gigi 4 5 2 1 2 6 ;

Figure 5–7. Data for the assignment model of Figure 5–6 (assign.dat).

To see why this makes sense, observe first that since there is only one student
available to be “shipped” from each “origin,” all of the Assign[s,p] variables
must be <= 1 as well as >= 0.

When also shipCost is replaced by assignRank, the terms of the objective
all have the form assignRank[s,p] * Assign[s,p], for some student s and
project p. This gives two possibilities for each term:

� If s is assigned to p, then Assign[s,p] is 1, in which case
the objective term assignRank[s,p] * Assign[s,p] takes the
value assignRank[s,p].

� If s is not assigned to p, then Assign[s,p] is 0, in which case
the objective term assignRank[s,p] * Assign[s,p] takes the
value 0.

When all these terms are added up, they include — besides a lot of zeroes —

the rank that each student gave to the project that student was assigned. This
is exactly the objective that we wish to have.

What if some of the variables come out to fractional values between 0 and 1?
The transportation model’s integrality property assures that we do not have to
worry about this possibility. There must exist an optimal solution in which the
variables all take integer values, which is this case means they must all take
values 0 or 1 since these are the only integers that are >= 0 and <= 1.

Figure 5–7 gives an example of data for our assignment model, with 19 stu-
dents, 6 companies, and project groups of 3 or 4. A solver for linear programs
readily obtains a result:

ampl: model assign.mod;
ampl: data assign.dat;

ampl: solve;
MINOS 5.5: optimal solution found.
73 iterations, objective 25

Compared to the lowest possible objective of 19 — when every student gets
their first choice — this looks to be pretty good. We do not want to have AMPL
display all of the variables’ values, however, since they include a huge number of
zeroes as well as some tiny magnitudes (such as 7.94749e-16 and -4.02805e-
16) that should be treated as zero. Instead we ask for a list of only the terms
assignRank[s,p] * Assign[s,p] that are nonzero:

ampl: option omit zero rows 1, display eps .000001;

ampl: display {s in STU, p in PRJ} assignRank[s,p]*Assign[s,p];

assignRank[s,p]*Assign[s,p] :=
Cane Gentil UOP 2
Chung Michael Abbott 1
Goldrich Matthew NMFF 1
Grasparil Michelle UOP 2
Henry Steve Zebra 2
Jurishica Carley NMFF 1
Laurie Alina UOP 2
Lawless Katie GM EMD 1
Morrow Timothy Zebra 1
Rincones Daianna Abbott 1
Shakoor Kashif Littelfuse 1
Sheth Vihar Abbott 1
Stewart Eric NMFF 1
Urquhart Geoff Zebra 2
Walton Caroline GM EMD 1
Welsch Todd GM EMD 1
Wible Barry Littelfuse 1
Yee Christine Littelfuse 2
Yuen Gigi NMFF 1

We see that everyone has received their first or second choice — a very good
assignment indeed.

5.5 Generalized and extended transportation models

In the Section 5.3 examples we assume that when a certain number of units
of material is shipped from origin o, the same amount of material arrives at
destination d. Often this is not quite the case, however; the amount arriving
is rather some fraction or multiple of the amount sent. A generalized trans-
portation problem takes such a conversion into account, and can be solved by
a corresponding generalized model.

As a specific illustration, we imagine that not all of the ore shipped from an
origin in the model of Figure 5–3 arrives at the destination; there is rather some
loss in transit. We handle this situation by defining a new AMPL parameter to
represent the fraction that is lost:

param shipLoss {ORIG,DEST} >= 0, <= 1;
# shipLoss[o,d] is fraction of material lost
# in shipment from origin o to destination d

We continue to interpret Ship[o,d] as the amount shipped out of origin o, so
that the constraints at the origins are as before:

subject to OrigAvl {o in ORIG}:
origLo[o] <= sum {d in DEST} Ship[o,d] <= origHi[o];

But now the amounts shipped into destinations are given by 1 - shipLoss[o,d]
times Ship[o,d]:

subject to DestReq {d in DEST}:
destHi[d] >=

sum {o in ORIG} (1 - shipLoss[o,d]) * Ship[o,d]
>= destLo[d];

When these additions and changes are made to the Figure 5–3 AMPL formulation
(transgen.mod), and the loss is set to 0.05 for all origin-destination pairs in the
Figure 5–5 AMPL data file (transgen.dat), the results are as follows:

ampl: model tranloss.mod;
ampl: data tranloss.dat;

ampl: solve;
MINOS 5.5: optimal solution found.
16 iterations, objective 2067.789474

ampl: option display 1col 0, display transpose -10;
ampl: option display precision 3;

ampl: display Ship;

Ship [*,*]
: DET FRA FRE LAF LAN STL WIN :=
CLEV 0 0 0 0 7 7 1
GARY 0 12 6.32 4.21 0 4.58 4.89
PITT 9.47 0.632 0 0 10.9 0 4.63

Here we no longer have an integer solution. The integrality property of “pure”
transportation problems evidently does not apply to generalized ones.

To get a solution in integral numbers of truckloads in this case, we must

specify integrality explicitly in the declaration of the Ship variables,

var Ship {o in ORIG, d in DEST}
integer, >= shipLo[o,d], <= shipHi[o,d];

after which we can switch to a solver that can optimize over integer solutions:

ampl: option solver cplex;
ampl: solve;

CPLEX 8.1.0: optimal integer solution; objective 2159
13 MIP simplex iterations
0 branch-and-bound nodes

ampl: display Ship;

Ship [*,*]
: DET FRA FRE LAF LAN STL WIN :=
CLEV 0 0 0 0 3 7 5
GARY 0 12 7 5 0 5 3
PITT 10 1 0 0 15 0 3

As is usually the case, the optimal integer solution does not derive in any com-
pletely obvious way from the optimal fractional one. Rounding all the values in
the fractional solution up to the next integer, for example, gives a solution that
meets all demands but that ships more from origin GARY than is available there.

Another common situation consists of a transportation problem plus some
further constraints to reflect complications of the application at hand. Often
the extra constraints interfere with the integrality property, even though they
are very simple in form.

We can illustrate using our assignment model. Supposed we have promised
to assign the following students to the same group: Steve Henry and Geoff
Urquhart; and Alina Laurie, Christine Yee, and Gigi Yuen. We can impose these
requests on the solution returned by our model, by adding three ad hoc con-
straints:

subject to SameTeam1 p in PRJ:
Assign["Henry Steve",p] = Assign["Urquhart Geoff",p];

subject to SameTeam2a p in PRJ:
Assign["Laurie Alina",p] = Assign["Yee Christine",p];

subject to SameTeam2b p in PRJ:
Assign["Yee Christine",p] = Assign["Yuen Gigi",p];

We add these to the model (assign2.mod) and re-solve:

ampl: model assign2.mod;
ampl: data assign.dat;

ampl: solve;
MINOS 5.5: optimal solution found.
58 iterations, objective 25

ampl: display eps .000001;
ampl: option display precision 3;

ampl: display Assign;

Assign [*,*]
: Abbott GM EMD Littelfuse NMFF UOP Zebra :=
Cane Gentil 0 0 0 1 0 0
Chung Michael 1 0 0 0 0 0
Goldrich Matthew 0 0 0 1 0 0
Grasparil Michelle 0 0 0 0 1 0
Henry Steve 0 0.5 0 0 0 0.5
Jurishica Carley 0 0 0 1 0 0
Laurie Alina 0 0 0.333 0 0.667 0
Lawless Katie 0 1 0 0 0 0
Morrow Timothy 0 0 0 0 0 1
Rincones Daianna 1 0 0 0 0 0
Shakoor Kashif 0 0 1 0 0 0
Sheth Vihar 1 0 0 0 0 0
Stewart Eric 0 0 0 1 0 0
Urquhart Geoff 0 0.5 0 0 0 0.5
Walton Caroline 0 0 0 0 0 1
Welsch Todd 0 1 0 0 0 0
Wible Barry 0 0 1 0 0 0
Yee Christine 0 0 0.333 0 0.667 0
Yuen Gigi 0 0 0.333 0 0.667 0

The objective remains at 25, but it is achieved by a fractional solution that
makes no sense when students are expected to devote all their time to one
project.

To find an integer solution here we again need to declare the variables as
integral and to invoke an appropriate solver:

ampl: redeclare var Assign {STU, PRJ} integer, >= 0, <= 1;

ampl: option solver cplex;
ampl: solve;

CPLEX 8.1.0: optimal integer solution; objective 26
30 MIP simplex iterations
0 branch-and-bound nodes

ampl: option omit zero rows 1;

ampl: display {s in STU, p in PRJ} assignRank[s,p]*Assign[s,p];

assignRank[s,p]*Assign[s,p] :=
Cane Gentil NMFF 1
Chung Michael Abbott 1
Goldrich Matthew Zebra 3
Grasparil Michelle NMFF 1
Henry Steve GM EMD 1
Jurishica Carley NMFF 1
Laurie Alina UOP 2
Lawless Katie GM EMD 1
Morrow Timothy Zebra 1
Rincones Daianna Abbott 1
Shakoor Kashif Littelfuse 1
Sheth Vihar Abbott 1
Stewart Eric NMFF 1
Urquhart Geoff GM EMD 1
Walton Caroline Zebra 2
Welsch Todd Littelfuse 2
Wible Barry Littelfuse 1
Yee Christine UOP 2
Yuen Gigi UOP 2 ;

We see that the same-team restrictions have forced the objective up to 26, and
caused many of the assignments to be rearranged. One student now receives
his third choice.

Is there any alternative solution in which every student gets their first or
second choice? We can answer this question by adding constraints that fix to
zero all variables representing choices worse than a certain “cutoff” value:

param cutoff > 0;

subject to Disallow {s in STU, p in PRJ: assignRank[s,p] > cutoff}:
Assign[s,p] = 0; ;

Adding this to our model to produce file assign3.mod, we can set the parameter
cutoff to 2 and re-solve as follows:

ampl: let cutoff := 2;
ampl: solve;

CPLEX 8.1.0: optimal integer solution; objective 26
13 MIP simplex iterations
0 branch-and-bound nodes

We see that the desired solution does exist, and has the same optimal objective
value of 26. This serves as a reminder that there can be more than one solution
to a problem that gives the same optimal objective value, and these objectives
may differ with respect to other properties of interest.

Can we give every student their first choice? When we try to answer this
question by setting cutoff to 1, we get a series of error messages:

ampl: let cutoff := 1;
ampl: solve;

presolve, variable Assign[’Cane Gentil’,’Zebra’]:
impossible deduced bounds: lower = 2, upper = 1;
difference = 1

presolve, variable Assign[’Cane Gentil’,’UOP’]:
impossible deduced bounds: lower = 3, upper = 1;
difference = 2

presolve, variable Assign[’Cane Gentil’,’NMFF’]:
impossible deduced bounds: lower = 0, upper = -3

presolve, variable Assign[’Cane Gentil’,’GM EMD’]:
impossible deduced bounds: lower = 0, upper = -1

presolve, constraint Disallow[’Cane Gentil’,’Littelfuse’]:
all variables eliminated, but upper bound = -1 < 0

Infeasible constraints determined by presolve. ;

Really only the last line is of importance here; it states that AMPL has deter-
mined, in its “presolve” phase — without even sending the generated instance
to the solver — that feasible solution to the constraints is possible. Thus giving
everyone their first choice is impossible, too.

6. Replications and Hybrids

Some of the models described in preceding sections can give rise to quite
large linear programs—ones that have hundreds or thousands of variables and
constraints. However, there is usually a natural limit to the number of inputs
and outputs or origins and destinations in a problem. A factory can use only so
many scarce resources and can make only so many products, for example.

The largest linear programs are seldom just large instances of the models
that you have already seen. Usually they are more complex linear programs that
have been built up in some way from simpler models. This section gives some
examples of ways in which complex models can be built. Most straightforwardly,
a model can be replicated over a new index set, creating multiple linear programs
that are then tied together into a single large one. Sections 6.1 and 6.2 present
replications over products and over times, respectively. Another approach is to
replicate each of two models over an index set in the other, after which they may
be combined into a larger hybrid model. Section 6.3 describes an example that
combines an output model replicated over origins with a transportation model
replicated over products.

6.1 A multicommodity transportation model

The transportation model of the previous chapter was concerned with ship-
ping a single product from origins to destinations. Suppose now that we are to
ship several. In the terminology of the AMPL model of Figure 5–3, we can define
a new set, PROD, whose members represent the different products, and we can
add PROD to the indexing of every component in the model.

The result can be seen in Figure 6–1. Because the parameters and variables
are indexed over one more set in this version, they have one more subscript:
origHi[o,p] for the amount of product p available to ship from origin o,
Ship[o,d,p] for the amount of p to be shipped from o to d, and so forth. The
total shipping cost to be minimized is summed over products as well as origins
and destinations, and there are constraints on the shipments of each product
at each origin and destination. To keep things simple we focus on the common
case where all of the origin lower bounds are zero and all of the destination
upper bounds are infinite, so that the parameters origLo and destHi from the
original transportation model can be dropped.

By itself, this replicated linear program is not too interesting. If you needed
to solve it for P products, you could break it up into P separate instances of
the single-product model, one instance for each product. To get the overall
minimum cost, you would do P separate solves of the Figure 5–3 model, each
with a different data file representing the transportation supplies, demands,
costs, and limits for a particular product. You would get the multi-commodity
minimum cost by adding up all the minimum costs for the individual products.

What makes a replicated linear program interesting is the addition of con-
straints or variables that tie together all the replicas. As an example, suppose
that instead of a limit on the amount of each product shipped along each route,

set ORIG; # origins
set DEST; # destinations
set PROD; # products

param origHi {ORIG,PROD} >= 0;
param destLo {DEST,PROD} >= 0;
param shipCost {ORIG,DEST,PROD} >= 0;

param shipLo {ORIG,DEST,PROD} >= 0;
param shipHi {o in ORIG, d in DEST, p in PROD} >= shipLo[o,d,p];

var Ship {o in ORIG, d in DEST, p in PROD}
>= shipLo[o,d,p], <= shipHi[o,d,p];

minimize TotalCost:
sum {o in ORIG} sum {d in DEST} sum {p in PROD}

shipCost[o,d,p] * Ship[o,d,p];

subject to OrigAvl {o in ORIG, p in PROD}:
sum {d in DEST} Ship[o,d,p] <= origHi[o,p];

subject to DestReq {d in DEST, p in PROD}:
sum {o in ORIG} Ship[o,d,p] >= destLo[d,p];

Figure 6–1. The AMPL model of Figure 5–3 replicated over products.

we have a limit on the total amount of all products shipped on each route. Then
we can replace the declarations of ShipLo and ShipHi in Figure 6–1 by

param shipLo {ORIG,DEST} >= 0;
param shipHi {o in ORIG, d in DEST} >= shipLo[o,d];

# Limit on total shipped from each origin to each destination

We declare variable Ship without any single-variable lower or upper limits,

var Ship {ORIG,DEST,PROD} >= 0;

and add a new collection of constraints that says, for each origin o and destina-
tion d, that the total shipments of all products from o to d must lie between the
specified limits. Provided the shipments are measured in the same units for all
products, the new constraints can be written as:

subject to ShipLim {o in ORIG, d in DEST}:
shipLo[o,d] <= sum {p in PROD} Ship[o,d,p] <= shipHi[o,d];

The complete model is shown in Figure 6–2. This model cannot be reduced to
a separate LP for each product, because each added constraint shipLo[o,d] <=
sum {p in PROD} Ship[o,d,p] <= shipHi[o,d] involves shipments of all the
products. In finding the minimum cost, the shipments of these products need to
be considered simultaneously; if you optimized the shipments of different prod-
ucts separately, there would be no way to ensure that the limits shipLo[o,d]
and shipHi[o,d] would be satisfied by the total volume of products shipped

set ORIG; # origins
set DEST; # destinations
set PROD; # products

param origHi {ORIG,PROD} >= 0;
param destLo {DEST,PROD} >= 0;
param shipCost {ORIG,DEST,PROD} >= 0;

param shipLo {ORIG,DEST} >= 0;
param shipHi {o in ORIG, d in DEST} >= shipLo[o,d];

var Ship {o in ORIG, d in DEST, p in PROD} >= 0;

minimize TotalCost:
sum {o in ORIG} sum {d in DEST} sum {p in PROD}

shipCost[o,d,p] * Ship[o,d,p];

subject to OrigAvl {o in ORIG, p in PROD}:
sum {d in DEST} Ship[o,d,p] <= origHi[o,p];

subject to DestReq {d in DEST, p in PROD}:
sum {o in ORIG} Ship[o,d,p] >= destLo[d,p];

subject to ShipLim {o in ORIG, d in DEST}:
shipLo[o,d] <= sum {p in PROD} Ship[o,d,p] <= shipHi[o,d];

Figure 6–2. A multicommodity transportation model (multi.mod).

from o to d.
This formulation is called a multicommodity transportation model. ForO ori-

gins, D destinations, and P products, there areODP variables andOP�DP�OD
constraints, so such a model can yield a rather large linear program even though
the numbers of origins, destinations and products are not especially large indi-
vidually. Moreover, the integrality property of single-commodity transportation
models does not carry over to the multicommodity case. Finding an optimal
integral solution to a multicommodity problem can be much more difficult than
finding an optimal fractional solution.

Although the costs, supplies and demands are all replicated over products
in this example, only some may need to be replicated in an application. For
example, the shipping rate per unit may be the same amount shipCost[o,d]
for any product sent from o to d. Then the objective would be sum {o in ORIG}
sum {d in DEST} sum {p in PROD} shipCost[o,d] * Ship[o,d,p].

6.2 A multi-period output model

Different kinds of linear programs result when models are replicated over
time rather than over products or places. The difference derives from the way
in which the replicas are tied together.

As an example, we consider the maximum-value product mix model from
Figure 4–3, replicated over periods of time. Individual periods can be of any
duration, but in planning models they are most often on the order of weeks,

set RAW; # raw materials
set PROD; # products

param T integer > 0;

param prodLo {PROD,1..T} >= 0; # lower bounds on products made
param prodHi {PROD,1..T} >= 0; # upper bounds on products made
param prodVal {PROD,1..T} >= 0; # values per ton of products made

param rawHi {RAW,1..T} >= 0; # raw material amounts available
param raw2p {RAW,PROD} >= 0;

# units of raw material to make a unit of product

var ProdMake {p in PROD,t in 1..T} >= prodLo[p,t], <= prodHi[p,t];
# units of each product to be made

maximize TotalValue:
sum {p in PROD, t in 1..T} prodVal[p,t] * ProdMake[p,t];

subject to RawLimits {r in RAW, t in 1..T}:
sum {p in PROD} raw2p[r,p] * ProdMake[p,t] <= rawHi[r,t];

Figure 6–3. The model of Figure 4–3 replicated over periods.

months, or years. We define a new AMPL parameter to represent the number of
periods,

param T integer > 0;

Then in the model we index parameters, variables, and constraints over 1..T,
which is the AMPL expression for the set of all integers from 1 through T. Only
parameter io is not indexed over periods; for purposes of this example we
imagine that the input-output coefficients can be regarded as fixed over time.
The full replicated model appears in Figure 6–3.

It would be possible to tie the replicas together as in the previous examples,
by imposing some constraints over all periods. However, it is usually more
realistic to add variables or constraints that simply relate each period to the
next one. The most common way of doing this is by adding inventories of certain
materials in the model.

For purposes of illustration, suppose that we expect to be able to obtain
a certain amount of each raw material in each period, and that we can hold
inventories of these materials for use in production in later periods. Then the
model requires a unit inventory cost for each raw material in each period, as
well as decision variables that stand for units of each raw material to be held in
inventory at each period:

param invCost {RAW,1..T} >= 0;
var InvHold {RAW,1..T} >= 0;

We adopt the convention that InvHold[r,t] represents the units of raw ma-
terial r in inventory at the end of period t, and hence also at the beginning of

period t+1.
The total inventory cost over all raw materials and T periods is sum {i in

RAW, t in 1..T} invCost[r,t] * InvHold[r,t], which must be subtracted
from the total value of the outputs to give the net total value:

maximize NetTotalValue:
sum {p in PROD, t in 1..T} prodVal[p,t] * ProdMake[p,t]

- sum {r in RAW, t in 1..T} invCost[r,t] * InvHold[r,t];

How should the constraints be modified to incorporate the inventories? For
each raw material r in each period t, we want the constraints to say that

material r allocated in period t � material r available in period t.

Raw material can be allocated in two ways: by using it in production, or by
putting it into inventory. The amount used in production is sum {p in PROD}
io[r,p] * ProdMake[p,t] as in Figure 6–3, while the amount put into inventory
is by definition InvHold[r,t]. Raw material can also be made available in two
ways: by obtaining it in the current period, or by taking it out of inventory left
from the previous period. The amount that can be obtained is rawHi[r,t],
while the inventory from the previous period is defined to be InvHold[r,t-1].

Substituting all these expressions, the desired form of the constraint is

subject to RawBalance {r in RAW, t in 1..T}:
sum {p in PROD} raw2p[r,p] * ProdMake[p,t] + InvHold[r,t]

<= rawHi[r,t] + InvHold[r,t-1];

This constraint replaces Figure 6–3’s constraint RawLimits on the inputs used in
each period. In fact RawBalance is a straightward generalization of RawLimits
with appropriate inventory quantities added to its left and right sides.

The structure of the new constraints changes the model significantly. Vari-
able InvHold["flex",2], for example, appears both on the left of the con-
straint RawBalance["flex",2], and on the right of RawBalance["flex",3].
The appearance of inventory variables in constraints for two successive periods
is what ties the replicated linear programs together. Intuitively, we can’t solve a
separate linear program for each period, because production in any one period
is potentially affected by inventories carried forward from previous periods, and
in turn potentially affects the inventories in later periods.

When we make all of the indicated changes and additions to the model, and
try to solve for a repesentative collection of data, AMPL signals an error in the
constraints:

ampl: model prodT.mod;
ampl: data prodT.dat;

ampl: solve;
Error executing "solve" command:
error processing constraint RawLimits[’drpu’,1]:

invalid subscript InvHold[’drpu’,0]

The problem here is that in a constraint for period t equal to 1, the second
subscript of InvHold[r,t-1] is zero — but we defined the variable InvHold to
be indexed only over 1..T.

When we change 1..T to 0..T in the declaration of InvHold, then the error
disappears:

ampl: redeclare var InvHold {RAW,0..T} >= 0;

ampl: solve;
MINOS 5.5: optimal solution found.
20 iterations, objective 1095420

As in our example at the end of Chapter 1, however, a closer look at the result
here shows that it makes no sense. All of the production amounts

ampl: option display 1col 0, display transpose -100;
ampl: display ProdMake;

ProdMake [*,*]
: 1 2 3 4 :=
2716 600 450 450 300
2717 400 300 350 300
2903 350 350 350 350

are at their upper limits,

ampl: display prodHi;

prodHi [*,*]
: 1 2 3 4 :=
2716 600 450 450 300
2717 400 300 350 300
2903 350 350 350 350

while the inventories start at huge values that are progressively drawn down
until they fall to zero at the last period:

ampl: display InvHold;

InvHold [*,*]
: 0 1 2 3 4 :=
drpu 280000 160000 90000 10000 0
flex 233000 201000 113000 18000 0

Now the problem is that while we’ve added variables InvHold[p,0] for each
p in PROD, we have not properly constrained them. InvHold[p,0] denotes the
amount of product p at the end of period 0, or equivalently at the beginning
of period 1 — the initial inventory. Because the inventory holding costs are
low relative to the product values, it is optimal to set the initial inventories
high enough to make up any difference between the greatest production allowed
(by the prodHi bounds) and the greatest that can be achieved (subject to the
RawLimits constraints). This explains both why the inventory amounts in the
solution are huge and why the production amounts all equal their limits.

Typically the initial inventories represent those currently on hand, and hence
are known quantities rather than variables to be decided. Hence the model is
readily fixed by defining a parameter to represent initial inventories,

param invInit {RAW} >= 0;

and by introducing a constraint to fix the initial inventory variables to the values
given by these parameters:

subject to InitialInventory {r in RAW}:
InvHold[r,0] = invInit[r];

With these changes a reasonable solution is produced, though of course with a
lower maximum objective value:

ampl: model prodT.mod;
ampl: data prodT.dat;

ampl: solve;
MINOS 5.5: optimal solution found.
13 iterations, objective 816760

set RAW; # raw materials
set PROD; # products

param T integer > 0; # number of periods

param prodLo {PROD,1..T} >= 0; # lower bounds on products made
param prodHi {PROD,1..T} >= 0; # upper bounds on products made
param prodVal {PROD,1..T} >= 0; # values per ton of products made

param invInit {RAW} >= 0; # initial raw material inventories
param invCost {RAW,1..T} >= 0; # raw material inventory costs
param rawHi {RAW,1..T} >= 0; # raw material amounts available
param raw2p {RAW,PROD} >= 0;

# units of raw material to make a unit of product

var ProdMake {p in PROD, t in 1..T} >= prodLo[p,t], <= prodHi[p,t];
# units of each product made in each period

var InvHold {RAW,0..T} >= 0;
# units of raw material held after each period

maximize NetTotalValue:
sum {p in PROD, t in 1..T} prodVal[p,t] * ProdMake[p,t]

- sum {r in RAW, t in 1..T} invCost[r,t] * InvHold[r,t];

subject to RawBalance {r in RAW, t in 1..T}:
sum {p in PROD} raw2p[r,p] * ProdMake[p,t] + InvHold[r,t]

<= rawHi[r,t] + InvHold[r,t-1];

subject to InitialInventory {r in RAW}:
InvHold[r,0] = invInit[r];

Figure 6–4. The model of Figure 6–3 with periods tied together by inventories and with
initial inventories as parameters.

ampl: display ProdMake;

ProdMake [*,*]
: 1 2 3 4 :=
2716 100 150 150 250
2717 400 150 350 300
2903 260 350 350 350

ampl: display InvHold;

InvHold [*,*]
: 0 1 2 3 4 :=
drpu 2000 0 20000 0 0
flex 6000 50000 43000 8000 0

The completed model is shown in Figure 6–4, and the problem instance that we
have been analyzing is given by the data in Figure 6–5.

The inventories at the end of the last period, InvHold[r,T] for r in RAW,
will be zero in every optimal solution, because any positive last-period inven-

set RAW := drpu flex ;
set PROD := 2716 2717 2903 ;

param T := 4 ;

param prodLo: 1 2 3 4 :=
2716 100 150 150 150
2717 50 25 50 25
2903 75 90 90 95 ;

param prodHi: 1 2 3 4 :=
2716 600 450 450 300
2717 400 300 350 300
2903 350 350 350 350 ;

param prodVal: 1 2 3 4 :=
2716 250 150 180 150
2717 300 250 300 200
2903 290 290 290 310 ;

param invInit := drpu 2000 flex 6000 ;

param rawHi: 1 2 3 4 :=
drpu 150000 150000 150000 180000
flex 200000 100000 100000 140000 ;

param invCost: 1 2 3 4 :=
drpu 0.03 0.03 0.03 0.03
flex 0.04 0.04 0.04 0.04 ;

param raw2p: 2716 2717 2903 :=
drpu 200 200 200
flex 200 140 160 ;

Figure 6–5. Data for the model of Figure 6–4, used by the example in the text.

tory would decrease the objective (due to inventory costs) without contributing
anything to the total value of products made. This is an example of an end effect:
a distortion in the solution due to the fact that the model ends abruptly after a
finite number of periods. End effects can sometimes be prevented by addition
of appropriate constraints; for example, the model could include some lower
limit on final inventories, if a reasonable value for the limit could be estimated.
Another approach is to use only the solution from earlier periods for actual
planning, with the later periods being retained in the model only to prevent end
effects in the earlier ones.

The same approach can be used to expand most kinds of single-period mod-
els into multi-period ones. In an input or output model, the inputs or outputs
(or both) can be inventoried. In a network model, the material being shipped can
be kept in inventories at either the origins or the destinations (or both). Various
hybrid and replicated models, like the input-output and multicommodity flow
examples, can also be further expanded into multi-period versions.

Multi-period linear programs easily become very large. In our example, if
there are R raw materials and P products, then the original output model has
R constraints in P variables, while the multi-period version has on the order of
RT constraints in �R � P�T variables. The number of constraints increases by a
factor of T , and the number of variables by an even larger factor.

6.3 A hybrid output/transportation model

Larger linear programs also arise by replicating different models and com-
bining the results in a way that ties all of the replicas together. As an example
we develop a hybrid that combines features of the output and transportation
models.

set RAW; # raw materials
set PROD; # products

set ORIG; # origins

param prodLo {ORIG,PROD} >= 0;
param prodHi {ORIG,PROD} >= 0;
param prodVal {ORIG,PROD} >= 0;

param rawHi {ORIG,RAW} >= 0;
param io {ORIG,RAW,PROD} >= 0;

var ProdMake {o in ORIG, p in PROD} >= prodLo[o,p], <= prodHi[o,p];

maximize TotalValue:
sum {o in ORIG} sum {p in PROD} prodVal[o,p] * ProdMake[o,p];

subject to RawLimits {o in ORIG, r in RAW}:
sum {p in PROD} io[o,r,p] * ProdMake[o,p] <= rawHi[o,r];

Figure 6–6. The output model of Figure 4–3, replicated over origins.

set RAW; # raw materials
set PROD; # products

set ORIG; # origins
set DEST; # destinations

param prodLo {ORIG,PROD} >= 0;
param prodHi {ORIG,PROD} >= 0;
param prodVal {ORIG,PROD} >= 0;

param rawHi {ORIG,RAW} >= 0;
param io {ORIG,RAW,PROD} >= 0;

param destLo {DEST,PROD} >= 0;
param shipCost {ORIG,DEST,PROD} >= 0;

var ProdMake {o in ORIG, p in PROD} >= prodLo[o,p], <= prodHi[o,p];

var Ship {o in ORIG, d in DEST, p in PROD} >= 0;

maximize NetTotalValue:
sum {o in ORIG} sum {p in PROD} prodVal[o,p] * ProdMake[o,p]

- sum {o in ORIG} sum {d in DEST} sum {p in PROD}
shipCost[o,d,p] * Ship[o,d,p];

subject to RawLimits {o in ORIG, r in RAW}:
sum {p in PROD} io[o,r,p] * ProdMake[o,p] <= rawHi[o,r];

subject to OrigAvl {o in ORIG, p in PROD}:
sum {d in DEST} Ship[o,d,p] <= ProdMake[o,p];

subject to DestReq {d in DEST, p in PROD}:
sum {o in ORIG} Ship[o,d,p] >= destLo[d,p];

Figure 6–7. The replicated transportation model of Figure 6–1 combined with the
replicated output model of Figure 6–6. Limits origHi of the transportation model are
replaced by the variables ProdMake of the output model.

We again start with a transportation model replicated over products, as in
Figure 6–1. But now we require the products to be made from raw materials at
each origin. To do so we again replicate the output model of Figure 4–3, but this
time over origins, as shown in Figure 6–6.

To combine these two models, we observe that the availabilites at the ori-
gins, origHi[o,p], must be consistent with the production levels at the origins,
ProdMake[o,p]. So we drop the origHi parameters, and replace origHi[o,p]
by ProdMake[o,p] in the OrigAvl constraints. For the objective we take the
total value from the replicated output model minus the total shipping cost from
the replicated transportation model. Figure 6–7 shows the result.

The OrigAvl and DestReq constraints in the hybrid model define a sepa-
rate transportation network for each product, but the RawLimits constraints
tie these networks together by limiting the raw materials consumed in making
all products at each origin. Another way to view the situation is to observe
that the RawLimits constraints define an independent production problem at

each origin, but the DestReq constraints unite these problems by imposing a
requirement on total shipments of each product to each destination.




